
1 EINLEITUNG

1 Einleitung

In dieser Bachelorarbeit soll das Computeralgebrasystem (CAS ) Maxima

anhand von Algorithmen aus der Computeralgebra vorgestellt werden. Da-

bei werden insgesamt zwei Schwerpunkte bedient. Einerseits die fachliche

Komponente, die wichtige Definitionen und Sätze aus der Algebra und Compu-

teralgebra widerspiegelt und deren Beweise oftmals bereits den Prototyp für

viele der behandelten Algorithmen liefern. Andererseits die anwendungsbe-

zogene Komponente, in der die Algorithmen in Maxima umgesetzt werden

und damit gleichermaßen als Einführung in die Funktionalität und Syntax

von Maxima dienen. Zusätzlich wird im Zuge dessen auch Codeanalyse be-

trieben und es werden Laufzeitvergleiche gemacht, um sukzessive effizientere

Algorithmen zu generieren.

Da die Algebra hinter den zu Grunde liegenden Problemstellungen der Algorith-

men oft sehr weitreichend ist und eine Reihe an Definitionen, Lemmata und

Sätzen mit sich bringt, werden gewisse Kenntnisse über Begriffe und Defi-

nitionen als gegeben vorausgesetzt. Ferner werden viele der Lemmata und

Sätze ohne Beweis als korrekt angenommen, um im Rahmen dieser Fachar-

beit möglichst viel Bezug auf die Arbeit mit Maxima nehmen zu können.

Weiterhin werden aufgrund des breit gefächerten Themengebiets einige Teil-

gebiete der Algebra stärker behandelt als andere. Insbesondere wird ein Fokus

auf die Zahlentheorie und Kryptographie gelegt, wobei die Polynomarithme-

tik nur exemplarisch angerissen wird.

Zur Struktur:

• Enthält ein Abschnitt einen Block, der mit Definition u. Satz

gekennzeichnet ist, so ist der
”
Satz“-Teil des Blocks jeweils in

Kursivschrift geschrieben.

• Textabschnitte, die direkt auf Definition, Lemmata, etc. folgen,

werden eingerückt, um eine deutlichere Abgrenzung zum Fließtext

zu gestalten.

2



2 ÜBER MAXIMA

2 Über Maxima

Maxima ist ein System zur Manipulation von symbolischen und numeri-

schen Ausdrücken. Diese beinhalten beispielsweise Ableiten, Integrieren, Dif-

ferentialgleichungen, Listen, Matrizen, u.V.m. Maxima bietet extrem präzise

numerische Resultate durch seine Implementierung der Ganz-/Gleitkomma-

zahlarithmetik und kann sowohl 2-dimensionale als auch 3-dimensionale Gra-

fiken erzeugen. Der Quellcode kann dabei auf vielen Betriebssystemen wie

Windows, Linux und MacOS X kompiliert werden. Somit ist Maxima auf

den gängigsten Betriebssystemen zugänglich. Maxima ist ein Nachfolger des

Computeralgebrasystems Macsyma, welches in den späten 1960er Jahren am

MIT entwickelt wurde. Es ist der einzige Nachfolger der noch frei zugäng-

lich ist und dank seiner Open-Source-Philosophie eine rege Mitwirkerschaft

besitzt. Der Ableger Maxima wurde von 1982 bis 2001 von William Schelter

weiterentwickelt und gewartet, welcher sich in 1998 die Rechte gesichert hat,

den Quellcode unter der GPL zu veröffentlichen. Maxima wird regelmäßig

geupdatet, um Fehler zu warten und Quellcode sowie Dokumentation weiter

zu verbessern.1

Maxima ist in Common Lisp programmiert, eine der ersten funktionalen Pro-

grammiersprachen. Dies spiegelt sich auch deutlich in der Syntax von Maxima

wieder.

Maxima kann als Konsolenanwendung genutzt werden (der sogenannte Text-

Mode), oder über das GUI-Tool wxMaxima, welches eine grafische, dokumen-

tenbasierte Nutzeroberfläche zur Verfügung stellt (welches auch in dieser

Facharbeit verwendet wird).

1Dieser Textabschnitt ist eine Zusammenfassung des Maxima-Abstracts, welcher auf
der offiziellen Maxima-Webseite www.maxima.sourcefourge.net zu finden ist.

3 EINFÜHRUNG IN MAXIMA

3 Einführung in Maxima

Um die Algorithmen später gut nachvollziehen zu können, soll zunächst ei-

ne kurze Einführung in die Syntax von Maxima gegeben werden. In den

folgenden Unterabschnitten wird die Funktionalität von Maxima nur grob

umrissen. Wichtige Befehle und Funktionen, welche für die Implementierung

der Algorithmen benötigt werden, werden in den jeweiligen Abschnitten be-

sprochen.

3.1 Grundbegriffe & Zuweisungen

Maxima kann in seiner Grundfunktion als CAS wie ein Taschenrechner be-

nutzt werden:

(%i1) 17 + 4 / 2;

(%o1) 19

Ein Semikolon (;) signalisiert Maxima dabei das Ende einer Zeile, bzw. das

Ende einer Anweisung. Jeder Eingabe/Ausgabe wird dabei automatisch ein

Label vergeben, sodass beide in späteren Programmblöcken separat ange-

sprochen werden können. Dies geschieht mit ’’Labelname:

(%i2) ’’%o1^3;

(%o2) 6859

Das (^) dient dabei als Potenzoperator und potenziert in diesem Fall das

Ergebnis (19) aus der ersten Rechnung mit 3.

Zuletzt berechnete Ergebnisse müssen nicht über das Label angesprochen

werden, sondern können direkt mit % abgerufen werden:

(%i3) sqrt(%), numer;

(%o3) 82.8190799272728

Mit ANWEISUNG, numer; erhält man das Ergebnis als Gleitkommazahl ge-

rundet. Die Genauigkeit lässt sich dabei mit fpprec: NACHKOMMASTELLEN

regulieren. Die Verwendung von numer ist oft wichtig zur Bewertung von Er-

gebnissen, da der Aufruf der Wurzelfunktion sqrt(6859) beispielsweise im

Normalfall 193/2 liefert.

3 EINFÜHRUNG IN MAXIMA

Zuweisungen werden in Maxima mit einem Doppelpunkt (:) ausgeführt, da

das Gleichheitszeichen (=) ausschließlich für Gleichungen reserviert ist:

(%i4) eq1 : x^2 + x - 2;

(%o4) x2 + x− 2

Das Polynom x2 + x− 2 ist nun in der Variable eq1 gespeichert und kann in

den folgenden Berechnungen verwendet werden.

(Achtung: Bei der Definition von Variablen dürfen keine bereits vorhandenen

Konstanten überschrieben werden, die z.B. durch %e für die eulersche Zahl

e oder %pi für die Kreiszahl π definiert sind.)

Mit:

(%i5) solve(eq1 = 0, x);

(%o5) [x = 1, x = -2]

lassen sich die Nullstellen der oben definierten Funktion berechnen. Der

solve-Befehl kann auch mit Listen als Argumenten aufgerufen werden:

solve([GLEICHUNG 1, . . . , GLEICHUNG n], [VAR 1, . . . , VAR k])

Dieser Aufruf versucht ein Gleichungssystem mit nGleichungen und k Variab-

len zu lösen. Listen können in Maxima eine beliebige Anzahl an verschiedenen

Datentypen enthalten, die durch ein Komma (,) getrennt sind.

Dass das erste Ergebnis aus %o5 tatsächlich stimmt, lässt sich in Maxima

leicht verifizieren:

(%i6) eq1, ’’%o5[1];

(%o6) 0

Da die Menge der Nullstellen eine Liste ist, kann mit LISTE[k] das k-te

Ergebnis angesprochen werden. Maxima benutzt dann das Ergebnis x = 1

als Ersetzungsregel für eq1.

(Achtung: Wie in vielen CAS beginnt Maxima bei mehrdimensionalen Ob-

jekten wie z.B. einer Liste bei 1 zu indizieren. Auch wenn diese Handhabung

in der mathematischen Software soweit üblich ist, beginnen Programmier-

sprachen im Normalfall die Indizierung bei 0.)

3 EINFÜHRUNG IN MAXIMA

3.2 Funktionsblöcke & Kontrollstrukturen

3.2.1 Funktionsblöcke

Um mehrzeilige Algorithmen mit Eingabeparametern in Maxima program-

mieren zu können, benötigen wir sogenannte Funktionsblöcke. Diese werden

definiert durch:

FUNKTIONSNAME(EINGABEPARAMETER) := block([LOKALE VARIABLEN],

ANWEISUNG 1, ANWEISUNG 2, . . . )

Im folgenden, einfachen Beispiel wird eine Funktion definiert, die 2 Zahlen

addiert und dann quadriert:

(%i7) add n square(x1, x2) := block([add, square],

add : x1 + x2,

print("Zahlen addiert!"),

square : add^2,

print("Ergebnis quadriert!"),

square

)$

Das Dollarzeichen ($) fungiert wie (;) und signalisiert das Ende einer Zeile/

Anweisung, unterdrückt dabei aber die Ausgabe. Dies ist sehr hilfreich bei

Funktionsblöcken, da Maxima den Block sonst nochmal komplett in der Aus-

gabe wiedergibt. Ein Aufruf unserer Funktion liefert folgendes Ergebnis:

(%i8) add n square(4,5);

Zahlen addiert!

Ergebnis quadriert!

(%o8) 81

Auch wenn die Kommentierung der Zwischenschritte durch die print(. . . )-

Befehle in diesem trivialen Beispiel nicht wirklich nötig ist, so kann die Aus-

gabe von Zwischenergebnissen/Ankerpunkten in komplexeren Algorithmen

deutlich von Vorteil sein.

(Achtung: Die print-Methode gibt zwar den Text als Zwischenschritt aus,

aber erst, nachdem die Berechnung vollständig durchgelaufen ist. Wir wer-

den später noch eine Methode kennenlernen, um Zwischenschritte während

der Berechnung auszugeben.)

3 EINFÜHRUNG IN MAXIMA

3.2.2 Kontrollstrukturen

Kontrollstrukturen stellen wichtige Funktionen zur Flusskontrolle eines Pro-

gramms dar, insbesondere if/else-Bedingungen und Schleifen.

Eine if/else-Bedingung:

if BEDINGUNG 1 then ANWEISUNG 1 elseif BEDINGUNG 2

then ANWEISUNG 2 elseif . . . else ANWEISUNG 0

evaluiert die erste ANWEISUNG k, für die BEDINGUNG k wahr ist. Sollte keine der

Bedingungen wahr sein, wird die else-Anweisung ausgeführt (ANWEISUNG 0).

Eine einfache Anwendung bietet die Prüfung, ob eine Zahl gerade ist:

(%i9) is even(n) := block([],

if mod(n,2) = 0 then

(print(n, "ist gerade"), NULL)

else

(print(n, "ist ungerade"), NULL)

)$

(%i10) is even(256);

256 ist gerade

(%o10) NULL

Da wir die Ausgabe des Ergebnisses bereits durch einen print-Befehl hand-

haben, definieren wir die Maxima-Ausgabe der Funktion als NULL, was in

unserem Beispiel eine erfolgreiche Terminierung des Programms bedeutet.

Wie oben gut zu sehen ist, müssen Anweisungen, die selbst wieder mehrere

Anweisungen enthalten, durch Kommata getrennt in eine Klammer einge-

schlossen werden (ANWEISUNG 1 1, ANWEISUNG 1 2, . . . ).

Maxima bietet zusätzlich eine Vielzahl an Schleifen und Sprunganweisun-

gen. Im nächsten Beispiel wird eine dieser Möglichkeiten vorgestellt, die

for-while-Schleife.

Das Programm soll die Summe der Quadrate aller ungeraden Zahlen bis zu

einer Grenze n bestimmen:

3 EINFÜHRUNG IN MAXIMA

(%i11) odd squares sum(n) := block([sum : 0],

for i : 1 step 2 while i <= n do

sum : sum + i^ 2,

sum

)$

(%i12) odd squares sum(15);

(%o12) 680

Die Schleife beginnt bei i = 1 und läuft solange, bis i > n ist. Dabei wird i

in 2-er-Schritten erhöht.

Weitere Kontrollstrukturen werden, wie bereits erwähnt, bei Bedarf in späte-

ren Abschnitten besprochen.

3.2.3 Rememberprogrammierung

Ein weiteres Feature findet sich in der Rememberprogrammierung. Wenn wir

in Maxima einen Funktionsblock aufrufen, so werden stets alle Berechnun-

gen neu ausgeführt. In manchen Situationen ist es aber hilfreich, wenn be-

reits berechnete Ergebnisse nicht nochmal berechnet werden müssen. Dies

bringt natürlich einen gewissen Speicheraufwand mit sich, ist aber für man-

che Anwendungen deutlich effizienter. Ein Paradebeispiel hierfür liefert die

rekursive Definition der Fibonacci-Zahlen:

(%i13) Fib[0] : 0$

(%i14) Fib[1] : 1$

(%i15) Fib[n] := Fib[n-1] + Fib[n-2]$

(%i16) Fib[10];

(%o16) 55

(%i17) Fib[15];

(%o17) 610

Beim Aufruf von Fib[10] wird die 10-te Fibonacci-Zahl berechnet. Gleich-

zeitig werden alle berechnen Fibonacci-Zahlen zwischen Fib[2] und Fib[10]

gespeichert. Beim Aufruf von Fib[15] werden dann nur noch die Zahlen zwi-

schen Fib[11] und Fib[15] neu berechnet, die anderen Ergebnisse/Berechnungen

werden aus dem Speicher genommen. Der Unterschied in der Art der Program-

mierung liegt in der Verwendung von eckigen Klammern ([]) anstelle von

runden Klammern (()) bei der Definition des Funktionsblocks.

4 LANGZAHLARITHMETIK

4 Langzahlarithmetik

Die Langzahlarithmetik beschäftigt sich mit Operationen und Algorithmen

für beliebig große ganze Zahlen (Z). Wir wollen uns in diesem Abschnitt zwei

Beispielen dieses Teilgebiets widmen, der Primfaktorzerlegung und dem (er-

weiterten) Euklidischen Algorithmus. Beide legen den Grundstein für spätere

Algorithmen der Modularen Arithmetik und der Kryptographie.

4.1 Teilertheorie & Primfaktorzerlegung

Um einen Algorithmus für die Primfaktorzerlegung definieren zu können,

muss die Existenz einer solchen zunächst gerechtfertigt werden. Beginnen

wir mit folgender Definition:

Definition 4.1. Seien x, y ∈ Z. Wir sagen
”
x teilt y“,

”
x ist Teiler von y“

oder
”
y ist ein Vielfaches von x“, falls ein d ∈ Z existiert, sodass

y = d · x ist. In diesem Fall schreiben wir x|y. Falls d, x 6= 1 sind, nennen wir

das Produkt d · x eine Zerlegung von y.

Mit der Definition eines Teilers können wir nun definieren, was es für eine

Zahl heißt, prim zu sein:

Definition 4.2. Eine Zahl p ∈ Z heißt prim, falls aus d ∈ Z mit d|p folgt:

Entweder ist d = 1 oder d = p. Dann nennen wir p eine Primzahl.

Umgangssprachlich ist eine Primzahl also nur durch 1 und sich selber teil-

bar.

Wir bezeichnen von nun an die Menge aller Primzahlen mit P.

Bevor wir uns der Primfaktorzerlegung zuwenden können, benötigen wir noch

folgenden Hilfssatz:

Lemma 4.3. Seien p ∈ P und a, b ∈ Z. Dann gilt:

p|a · b⇒ p|a oder p|b

Mit allen Hilfsmitteln ausgestattet können wir uns nun einem der wichtigs-

ten Sätze der Zahlentheorie annehmen:

4 LANGZAHLARITHMETIK

Satz 4.4. (Fundamentalsatz der Zahlentheorie) Jede Zahl x ∈ N≥2
besitzt eine eindeutige Zerlegung in Primzahlen (Primfaktorzerlegung).

Beweis (PFZ := Primfaktorzerlegung/-en)

Existenz: Beweis durch vollständige Induktion:

IA Sei x = 2. Da 2 Primzahl ist, hält die Vermutung.

IV Jede Zahl z ∈ N≥2 mit z < x besitzt eine PFZ.

IS Sei x > 2. Dann existieren für x zwei Möglichkeiten:

Möglichkeit 1: x ist prim. Dann stellt x sich selber dar und ist somit seine

eigene PFZ.

Möglichkeit 2: x besitzt eine Zerlegung x = a · b (a, b ∈ N≥2). Dann ist per

Definition einer Zerlegung aber a < x und b < x. Nach der IV existieren für

a und b jeweils eine PFZ, also auch für x = a · b als Produkt der einzelnen

PFZ.

Eindeutigkeit: Beweis durch vollständige Induktion:

IA Sei x = 2. Dann ist die PFZ trivialerweise eindeutig.

IV Für jede Zahl z ∈ N≥2 mit z < x ist die PFZ eindeutig.

IS Sei x > 2. Seien x = p1 · · · pn = q1 · · · qm mit pj ∈ P (j = 1, ..., n) und

qj ∈ P (j = 1, ...,m). Da p1|x = q1 · · · qm folgt mit Lemma 4.3, dass p1|qj für

ein j aus 1, ...,m. Da qj prim ist, folgt p1 = qj. Also ist x
p1
∈ N mit:

x > x
p1

= p2 · · · pn = q1 · · · qj−1 · qj+1 · · · qm

Aus der IV folgt, dass die PFZ von x
p1

gleich sein müssen, d.h.

p2, ..., pn = q1, ..., qj−1, qj+1, ..., qm bis auf Umordnung. Da aber p1 = qjist,

gilt auch p1, ..., pn = q1, ..., qm bis auf Umordnung.

Maxima liefert mit ifactors(n) (kurz für: Integerfactors) eine effiziente Imple-

mentierung zur Bestimmung der PFZ von n:

(%i18) next prime(10^20) · next prime(10^21);

(%o18) 100000000000000000050700000000000000004563

(%i19) ifactors(%);

(%o19) [[100000000000000000039,1],[1000000000000000000117,1]]

Die Funktion next prime(k) liefert dabei die nächste Primzahl größer k.

Dies gibt uns eine einfache Möglichkeit, die Funktion ifactors in ihrem

10



4 LANGZAHLARITHMETIK

worst-case zu testen, nämlich genau dann, wenn die Zahl zwei große Prim-

faktoren besitzt. Dies zeigt auch folgende Analyse:

(%i20) time(%o19);

(%o20) [30.161]

Der Befehl time(%o1, %o2, ...) liefert die Zeit in Sekunden, welche die

jeweiligen Ausgaben benötigt haben. In der Größenordnung einer 41-stelligen

Zahl benötigt der Algorithmus im worst-case-Szenario bereits ca. 30 Se-

kunden. Von der Komplexität des Faktorisierens profitieren insbesondere

Anwendungen der Kryptographie, wie wir in den späteren Abschnitten noch

sehen werden.

(Achtung: Bei der Arbeit mit wxMaxima steht die Arbeit in Form eines

kompletten Dokumentes im Vordergrund. Ändert man in diesem Prozess ei-

ne bereits ausgeführte Zeile ab oder schreibt diese neu, ändern sich auch %i

& %o-Nummer. Dies kann schnell zu Fehlern/falschen Ergebnissen führen,

insbesondere bei Befehlen, die bestimmte Ausgaben referenzieren.)

Wir werden uns dem Problem des Faktorisierens in einem späteren Kapital

nochmal annehmen, nachdem wir etwas Vorarbeit in der Modularen Arith-

metik geleistet haben. Bis dahin soll folgender Beispielalgorithmus gegeben

werden, um die Primfaktorzerlegung händisch zu implementieren:

(%i21) factor int(n) := block([div : 2, factors : [],

k : n, n new],

while k > 1 and not(primep(k)) do (

while integerp(n new : k / div) do (

factors : append(factors, [div]),

k : n new

),

div : next prime(div)

),

if primep(k) then factors : append(factors, [k]),

factors

)$

Dieser Algorithmus verwendet zwei interessante Schlüsselwörter, die wir et-

was näher betrachten wollen:

integerp(n)/primep(n): Test, ob n ∈ Z/n ∈ P ist.

append(liste 1, . . . , liste n): Fügt die Listen 1, . . . , n zusammen.

12



4 LANGZAHLARITHMETIK

Wie im Algorithmus gut zu sehen ist, kann die while-Schleife auch losgelöst

von der for-Schleife implementiert werden. Mehrere Bedingungen können da-

bei mit and/or zusammengefügt werden. not(BEDINGUNG) negiert die jewei-

lige BEDINGUNG.

Zwar stellt die Abfrage not(primep(k)) in Zeile 3 eine Optimierung zur

klassischen Probedivision2 dar, es zeigen sich aber doch deutliche Unterschie-

de zur effizienten Implementierung von Maxima :

(%i22) factor int(next prime(10^7) · next prime(10^8));

(%o22) [10000019,100000007]

(%i23) time(%o22);

(%o23) [25.5]

Auch wenn unser Algorithmus Zahlen der Größenordnung 10ˆ15 im worst-

case-Szenario noch in angemessener Zeit faktorisieren kann, so gilt dies nicht

für die Eingabe mit der Zahl aus unserem Initialbeispiel (%i18):

(%i24) factor int(’’%o18); /· Sollte abgebrochen werden ·/
Schon bei Zahlen der Größenordnung 1020 beträgt die Laufzeit des Algorith-

mus weit über 5 Minuten.

Die Eingabe aus %i24 sollte mit der Tastenkombination Strg+g abgebrochen

werden. Wie oben zu sehen ist, lassen sich mit /* Kommentar */ Kommen-

tare einfügen. (Maxima macht dabei automatisch aus (*) ein (·))
Nach dem Fundamentalsatz der Zahlentheorie wollen wir uns einem weiteren,

sehr wichtigen zahlentheoretischen Algorithmus widmen:

4.2 Der Euklidische Algorithmus

Wie in Definition 4.1 festgelegt, nennen wir eine Zahl x einen Teiler von y

nur dann, falls ein d existiert, sodass y = d · x (∈ Z). Diese Definition ist

sinnvoll, da jede Lockerung dieser Bedingung dazu führen würde, dass wir

den Bereich der ganzen Zahlen (Z) verlassen müssten. Trotzdem können wir

auch in Z eine Division y
x

mit x - y durchführen, in dem wir uns trivialerweise

der Grundschulmathematik bedienen:

2Die Probedivision beschreibt das Verfahren, in dem n sukzessive durch die Primzahlen
p ∈ P beginnend mit p = 2 dividiert wird, bis n = 1 ist.

4 LANGZAHLARITHMETIK

Definition u. Satz 4.5. (Division mit Rest) Seien y ∈ N≥0 und x ∈ N.

Dann gibt es genau ein (q, r) ∈ N≥0 × N≥0 und 0 ≤ r < x sodass:

y = q · x+ r

Wir nennen q = quotient(y, x) den ganzzahligen Quotient/Anteil von y

und x, sowie r = mod(y, x) den (Divisions-)rest der Division von y durch

x. Falls r = 0 ist, so gilt nach Definition 4.1 x|y.

Satz 4.5 lässt sich dabei recht intuitiv auf den Bereich der ganzen Zahlen

(y, x ∈ Z, x 6= 0) erweitern:

y < 0 & x > 0: Setze y = −y → Berechne y = q · x+ r →
Setze q = −q − 1 und r = −r + x

y > 0 & x < 0: Setze x = −x→ Berechne y = q · x+ r →
Setze q = −q

y < 0 & x < 0: Setze y = −y und x = −x→ Berechne y = q · x+ r →
Setze q = −q − 1 und r = −r + (−x)

Dabei wird stets die Bedingung 0 ≤ r < x beibehalten.

In Maxima liefern folgende Eingaben den Quotienten resp. den Rest:

(%i25) floor(2300/7);

(%o25) 328

(%i26) mod(2300, 7);

(%o26) 4

Man beachte dabei, dass die floor-Funktion nicht explizit den ganzzahli-

gen Quotienten der Division mit Rest berechnet, sondern die Rechnung in Q
durchführt und das Ergebnis auf die nächste ganze Zahl abrundet. Dabei gilt

aber stets quotient(y,x) = floor(y/x) := b y
x
c. Um den Euklidischen Algo-

rithmus definieren zu können, müssen wir zunächst definieren, was sein Ziel

ist:

Definition 4.6. Seien a, b, c ∈ Z. Die Zahl c heißt gemeinsamer Teiler

von a und b, falls c|a und c|b. Falls für c nur c ∈ {−1, 1} die Bedingung

erfüllt, nennen wir a und b teilerfremd. Falls für alle anderen Teiler d ∈ Z
mit d|a und d|b gilt d|c, heißt c der größte gemeinsame Teiler von a und

b

(c := gcd(a, b)).

4 LANGZAHLARITHMETIK

Die Definition des größten gemeinsamen Teilers führt uns zu folgendem

Algorithmus:

Definition u. Satz 4.7. Seien a, b ∈ Z. Das Iterationsschema:

xk−1 = qk · xk + rk

mit den Anfangsbedingungen (x0 = a, x1 = b) und den Iterationsvorschriften:

qk = quotient(xk−1, xk) rk = mod(xk−1, xk) xk+1 = rk

heißt der Euklidische Algorithmus. Dabei gilt die Abbruchbedingung

rk = 0.

Der oben beschriebene Algorithmus terminiert für alle a, b und liefert bei

n Schritten3 xn = gcd(a, b).

Um den Satz aus 4.7 zu beweisen, müssen wir davon ausgehen können,

dass der größte gemeinsame Teiler von a und b unter dem Iterationsschema

invariant bleibt. Diese Annahme führt zu folgendem Hilfssatz:

Lemma 4.8. Seien a, b ∈ Z mit a = q · b+ r durch Division mit Rest. Dann

gilt:

gcd(a, b) = gcd(b, r)

Beweis zu 4.7 Da der Euklidische Algorithmus mit der Division mit Rest

arbeitet, und für jeden Rest rk: 0 ≤ rk < xk gilt, muss nach endlich vielen

Schritten rk = 0 erreicht werden. Sei n der letzte Iterationsschritt. Aus Lem-

ma 4.8 folgt, dass gcd(a, b) = gcd(xn, rn = 0) sein muss. Aus der Definition

des größten gemeinsamen Teilers macht man sich schnell klar, dass für alle

x ∈ Z : gcd(x, 0) = x gelten muss, da stets x|0 gilt und x somit die größte

Zahl ist, die auch sich selber noch teilt. Daraus folgt:

gcd(a, b) = gcd(xn, 0) = xn

Die Konstruktion des Euklidischen Algorithmus liefert uns dabei zusätzlich

eine Identität für den größten gemeinsamen Teiler:

Da gcd(a, b) = xn = rn−1 und rn−1 = xn−2− qn−1 · xn−1 ist, gilt insbesondere

gcd(a, b) = xn−2 − qn−1 · xn−1. Durch sukzessive Substitution erhält man

so eine Darstellung gcd(a, b) = s · a + t · b mit s, t ∈ Z, die sogenannten

Bézoutkoeffizienten.
3Die Anzahl der Schritte wird auch euklidische Länge von a und b genannt.

4 LANGZAHLARITHMETIK

Der Euklidische Algorithmus samt Darstellung des Ergebnisses durch die

Bézoutkoeffizienten wird auch der erweiterte Euklidische Algorithmus

genannt. Diesen wollen wir in Maxima implementieren:

(%i27) ext gcd(a,b) := block([q, r, s : [1,0], swapped : false,

t : [0,1], tmp],

if a < b then (

tmp : a, a : b, b : tmp, swapped : true

),

if mod(a,b) = 0 then

if swapped then

return([b,[1,0]]) else

return([b,[0,1]]),

while not(mod(a,b) = 0) do (

[q,r] : [floor(a/b),mod(a,b)],

[a,b] : [b,r],

s : [s[2],s[1] - q · s[2]],
t : [t[2],t[1] - q · t[2]]

),

if swapped then

[b,[t[2],s[2]]] else [b,[s[2],t[2]]]

)$

Der Algorithmus überprüft zunächst, ob a < b gilt. Falls dies nicht der Fall

ist, werden die Zahlen getauscht und der Algorithmus arbeitet normal weiter.

Damit aber am Ende das richtige Ergebnis für die Zahlen a, b in Eingabe-

reihenfolge ausgegeben werden kann, wird mit der Variable swapped festge-

halten, ob getauscht wurde oder nicht. Falls b|a gilt, haben wir den gcd(a, b)

bereits gefunden, nämlich b, mit gcd(a, b) = 0 · a + 1 · b. Hier wird zum ers-

ten Mal das return(...)-Statement benutzt, welches das gesuchte Ergebnis

zurückliefert und gleichzeitig den Funktionsblock an dieser Stelle beendet.

(return(...) innerhalb einer Schleife beendet nur die Schleife!)

In der while-Schleife sieht man eine Technik, die auch im weiteren Verlauf

dieser Arbeit noch öfter Anwendung finden wird:

Mit [VAR 1, ..., VAR n] : [VAL 1, ..., VAL n] lassen sich mehrere Zu-

weisungen gleichzeitig machen.

4 LANGZAHLARITHMETIK

Eine weitere Eigenschaft von Maxima findet sich in der Arbeitsweise des Al-

gorithmus. Betrachten wir folgenden Aufruf:

(%i28) [a,b] : [234,12]$

(%i29) ext gcd(a,b);

(%o29) [6, [1, -19]]

(%i30) [a,b];

(%o30) [234,12]

In der Implementierung des Algorithmus ist zu sehen, dass die Werte a und

b nicht in lokalen Variablen zwischengespeichert werden, sondern direkt im

Algorithmus verwendet und verändert werden. Auf globaler Ebene bleiben

die Variablen a und b dennoch unverändert. Maxima verwendet also entweder

eine lokale Kopie der Eingabeparameter oder ändert die Werte am Ende des

Funktionsaufrufs wieder auf ihr Original. Bei Mathematica beispielsweise ist

das Verändern der Eingabeparameter eines Funktionsblocks innerhalb eines

solchen nicht erlaubt!

Maxima selber bietet natürlich auch eine Implementierung beider Varianten

des Euklidischen Algorithmus. Um die Suche nach den Funktionsnamen et-

was einfacher zu gestalten, bietet Maxima folgende Hilfestellung:

(%i31) apropos("gcd");

(%o31) [ef gcd,ef gcdex,ezgcd,gcd,gcdex,gf gcd,gf gcdex,gcd]

apropos("STICHWORT") liefert alle Funktionsnamen und Flags (Einstellun-

gen) die STICHWORT beinhalten4. Für eine genauere Beschreibung der Funk-

tion kann man ?? FUNKTIONSNAME oder describe("FUNKTIONSNAME") auf-

rufen. Folgender Aufruf ist somit äquivalent zu unserem Algorithmus:

(%i32) gcdex(a,b);

(%o32) [1,-19,6]

Hierbei sind die ersten beiden Argumente die Bézoutkoeffizienten und das

dritte Argument ist der größte gemeinsame Teiler.

4Deswegen kommt gcd hier auch zwei mal vor. Die Funktion selber und die Flag welche
steuert, ob der größte gemeinsame Teiler bei Umwandlungen mitgezogen werden soll. Das
Setzen dieser Flag auf false kann gewisse Prozesse beschleunigen, in denen der gcd nicht
weiter von Nöten ist.

4 LANGZAHLARITHMETIK

Ein Laufzeitvergleich der beiden Algorithmen macht hier wenig Sinn:

(%i33) seed : make random state(3993)$

(%i34) set random state(seed)$

(%i35) showtime : true$

Evaluation took 0.0000 seconds (0.0000 elapsed) using

0 bytes.

(%i36) gcdex(random(10^500), random(10^500));

Evaluation took 0.0270 seconds (0.0350 elapsed) using

3.622 MB.

(%o36) [105615214096097900641466857438[440 Ziffern]

189453297140473772094144467045,

-360244168306652521232854578100[440 Ziffern]

255727470295515905228882708693,1]

(%i37) set random state(seed)$

(%i38) ext gcd(random(10^500), random(10^500));

Evaluation took 0.1580 seconds (0.1600 elapsed) using

10.985 MB.

(%o38) [1,[105615214096097900641466857438[440 Ziffern]

189453297140473772094144467045,

-360244168306652521232854578100[440 Ziffern]

255727470295515905228882708693]]

Wie im Beispiel gut zu sehen ist, ist unser Algorithmus zwar etwas lang-

samer und speicheraufwendiger, evaluiert das Ergebnis aber trotzdem sehr

schnell, zu mal es sich um Zahlen mit ca. 500 Stellen handelt.5

Die Funktionen make random state/set random state sorgen dafür, dass

wir die von random(GRÖßE) erzeugten Zufallszahlen reproduzieren können.

make random state(ZAHL) erstellt dabei einen Startpunkt für den

Random-Number-Generator (RNG) von Maxima , welcher zu ZAHL eindeu-

tig zugeordnet ist. set random state(RANDOM STATE) setzt den RNG immer

wieder auf den von uns gewählten Ausgangspunkt zurück.6

5Die random-Funktion schwankt dabei bei der Erstellung der Zahlen zwischen ∼498-500
Stellen.

6Wann der RNG voranschreitet, ist von Programm zu Programm unterschiedlich.

5 MODULARE ARITHMETIK

Die Flag showtime sorgt auf true dafür, dass jeweils Zeit-und Speicherver-

brauch mit ausgegeben werden. Diese kann nach der Evaluation wieder auf

false gesetzt werden:

(%i39) showtime : false$

5 Modulare Arithmetik

So simpel wie die Idee der Division mit Rest auch scheint, eröffnet sie uns

doch ein komplett neues Teilgebiet der Algebra/Zahlentheorie, das Gebiet

der Modularen Arithmetik. Damit wir uns den interessanten Aussagen und

Algorithmen dieses Gebiets widmen können, ist aber erstmal etwas algebra-

ische Vorarbeit nötig:

5.1 Gruppen, Ringe & Kongruenzen

Definition 5.1. Eine Menge G mit einer inneren, zweistelligen Verknüpfung

? (Abbildung ? : G × G → G, (a, b) 7→ a ? b) heißt Gruppe, falls die Ver-

knüpfung folgende Eigenschaften erfüllt:

1. (a ? b) ? c = a ? (b ? c) für alle a, b, c ∈ G (Assoziativität)

2. ∃e ∈ G mit a ? e = e ? a = e für alle a ∈ G (neutrales Element)

3. ∀a ∈ G ∃a−1 ∈ G mit a ? a−1 = a−1 ? a = e (inverses Element)

Gilt zusätzlich a ? b = b ? a ∀a, b ∈ G nennen wir (G, ?) abelsche Gruppe.

Erfüllt ? nur die erste Bedingung, nennen wir (G, ?) Halbgruppe.

Dies führt uns zur Definition eines Rings:

Definition 5.2. Eine Menge R mit zwei inneren, zweistelligen Verknüpfun-

gen +, · heißt Ring, falls folgende Eigenschaften erfüllt sind:

1. (R,+) ist abelsche Gruppe

2. (R, ·) ist Halbgruppe

3. Es gelten die Distributivgesetze:

a · (b+ c) = (a · b) + (a · c) und

(a+ b) · c = (a · c) + (b · c) für alle a, b, c ∈ R

Besitzt (R, ·) ein neutrales Element, so nennen wir (R,+, ·) einen unitären

Ring (Ring mit Eins).

Ist (R, ·) abelsch, nennen wir R einen kommutativen Ring.

Offensichtlich bietet unsere bisher betrachtete Menge Z mit (Z,+, ·) ein

perfektes Beispiel für einen unitären, kommutativen Ring. Die Definition ei-

nes Rings ist essentiell für viele Teilgebiete der Algebra. Können wir zeigen,

dass eine Struktur ein Ring ist, so können wir uns dessen Ringeigenschaften

zu Nutze machen und darauf aufbauen.

Eine weitere Klassifizierung von wichtigen Elementen in Ringen liefert fol-

gende Definition:

Definition u. Satz 5.3. Sei (R,+, ·) ein unitärer, kommutativer Ring und

1R das Einselement (neutrale Element bzgl. ·). Ein Element a ∈ R heißt

Einheit, falls ein b ∈ R existiert mit:

a · b = b · a = 1R

Die Menge der Einheiten R× := {x ∈ R |x ist Einheit} bildet eine Gruppe.

Die Menge R× sind also alle Elemente, die ein Inverses bzgl. · besitzen.

Für unseren Beispielring (Z,+, ·) gilt somit Z× = {1,−1}.
Ausgehenden von unseren Definitionen wollen wir zunächst die Division mit

Rest in eine algebraische Struktur überführen:

Definition u. Satz 5.4. Zwei Zahlen a, b ∈ Z heißen kongruent modulo

p ∈ N≥2, falls sie bei der Division mit Rest durch p denselben Rest haben,

d.h.:

mod(a, p) = mod(b, p)

Wir schreiben dann a ≡ b (mod p).

Die Kongruenz (≡) definiert eine Äquivalenzrelation auf Z, d.h. sie liefert

eine Klasseneinteilung von Z in p Klassen ([0]p, · · · , [p− 1]p), wobei gilt:

a ∈ Z, r ∈ {0, · · · , p− 1} : a ∈ [r]p ⇔ mod(a, p) = r

Zusätzlich ist sie verträglich mit der Addition und (Skalar-)Multiplikation.

Die Äquivalenzklassen [a]p werden Restklassen genannt, mit ihrem zu-

gehörigen Vertreter a ∈ {0, · · · , p − 1} und Zp := {[0]p, · · · , [p − 1]p} als

Menge aller Restklassen.

Die Verträglichkeitseigenschaften aus Satz 5.4 liefern uns dabei folgende,

recht intuitive, Rechenregeln für Restklassen:

Definition u. Satz 5.5. Sei p ∈ N≥2 und ϕ Abbildung mit:

ϕ :Z → Zp
a 7→ ϕ(a) := [a]p

Damit definieren wir:

[a]p ⊕ [b]p := ϕ(a+ b) = [a+ b]p und

[a]p ⊗ [b]p := ϕ(a · b) = [a · b]p

als Addition bzw. Multiplikation in Zp. Außerdem gilt:

(Zp,⊕,⊗) ist unitärer, kommutativer Ring.

Mit Satz 5.5 können wir uns die Ringeigenschaften zu Nutze machen und

gleichzeitig eine interessante Möglichkeit zur Berechnung der Einheitengrup-

pe Z×p liefern:

Definition u. Satz 5.6. Ein Element a ∈ Zp ist genau dann eine Einheit in

(Zp,⊕,⊗), falls gcd(a, p) = 1 ist.

Daraus folgt: Ist p prim, so ist Zp Körper.

Das Inverse b = a−1 (mod p) wird modulares Inverses von a genannt, falls

a · b ≡ 1 (mod p) ist.

Das modulare Inverse lässt sich direkt über den Erweiterten Euklidischen

Algorithmus von a und p bestimmen:

(%i40) [s,t,g] : gcdex(3,7);

(%o40) [-2,1,1];

(%i41) mod(s,7);

(%o41) 5

Das modulare Inverse zu 3 in Z7 ist demnach 5. Dies ist offensichtlich korrekt,

wie die Rechnung 3 · 5 = 15 ≡ 1 (mod 7) zeigt. Maxima liefert aber auch eine

eigene Methode zum Bestimmen des modularen Inverses mit:

(%i42) inv mod(3,7);

(%o42) 5

Die Methoden der modularen Arithmetik lassen uns interessante Problem-

stellungen lösen:

5.2 Der chinesische Restsatz

Wir betrachten folgendes Rätsel: Ein Gruppe von Menschen kann so in 5-er

Reihen aufgestellt werden, dass eine 2-er Reihe übrigbleibt und gleichzeitig

so in 7-er Reihen aufgestellt werden, dass eine 4-er Reihe übrigbleibt. Wie

viele Personen befinden sich in der Gruppe?

Wie oft bei solchen Zahlenrätseln lassen sich die einzelnen Bedingungen in

Gleichungen ausdrücken. Die modulare Arithmetik liefert uns das Handwerks-

zeug dafür:

Ist x die Anzahl der Personen in der Gruppe, so lassen sich die obigen Be-

dingungen wie folgt darstellen:

x ≡ 2 (mod 5) und

x ≡ 4 (mod 7)

In unserem Beispiel lässt sich die Lösung recht einfach erraten. Besteht die

Gruppe aus 32 Personen, so sind beide Bedingungen erfüllt und das Rätsel

ist gelöst. Unsere Lösung ist aber keinesfalls eindeutig, da zum Beispiel eine

Gruppe von 67 Personen (67 ≡ 32 (mod 35)) das Rätsel gleichermaßen löst.

Diese Grundüberlegungen führen zu folgendem Satz:

Satz 5.7. (Chinesischer Restsatz) Seien p1, · · · , pn ∈ Z paarweise teiler-

fremd (gcd(pi, pk) = 1 für i 6= k) und a1, · · · , an ∈ Z. Dann besitzt das

Gleichungssystem:

x ≡ a1 (mod p1)

x ≡ a2 (mod p2)
...

x ≡ an (mod pn)

eine bis auf additive Vielfache von p1 · · · pn eindeutige Lösung.

Beweis

Wir konstruieren zunächst eine Lösung für die ersten beiden Gleichungen:

Da gcd(p1, p2) = 1 ist, existieren Bézoutkoeffizienten s, t ∈ Z mit:

s · p1 + t · p2 = 1

Definieren wir l := a2 · s · p1 + a1 · t · p2, so gilt:

l ≡ a1 · t · p2 ≡ a1 · (1− s · p1) ≡ a1 (mod p1)

l ≡ a2 · s · p1 ≡ a2 · (1− t · p2) ≡ a2 (mod p2)

Somit ist l Lösung für die ersten beiden Gleichungen. Da aber alle pi paar-

weise teilerfremd sind, sind insbesondere alle pj für j = 3, . . . , n teilerfremd

zu p1 · p2. Somit können die ersten beiden Gleichungen im Gleichungssystem

durch:

x ≡ l (mod p1 · p2)

ersetzt werden. Dadurch kann das Gleichungssystem sukzessive verringert

werden, bis aus den verbleibenden 2 Gleichungen eine finale Lösung l∗ mit

x ≡ l∗ (mod p1 · · · pn) gewonnen werden kann.

Eindeutigkeit: Seien l∗ und l̂∗ zwei finale Lösungen. Dann gilt:

l∗ − l̂∗ ≡ 0 (mod p1), . . . , l
∗ − l̂∗ ≡ 0 (mod pn)

also auch l∗ − l̂∗ ≡ 0 (mod p1 · · · pn).

Anhand der Konstruktionsvorschrift aus dem Beweis wollen wir den chinesisch-

en Restsatz in Maxima programmieren. Aufgrund der Art der Konstruktion

bietet sich hier eine rekursive Funktion an:

(%i43) chinese remainder(r,p) := block([g,s,t],

[s,t,g] : gcdex(p[1],p[2]),

if is(length(r) = 2) then

mod(r[1]·p[2] + r[2]·s·p[1],p[1]·p[2]) else

chinese remainder(

flatten([mod(r[1]·t·p[2] + r[2]·s·p[1],p[1]·p[2]),
lastn(r, length(r)-2)]),

flatten([p[1]·p[2], lastn(p, length(p)-2)])

)

)$

In diesem Algorithmus finden sich 2 neue Funktionen. flatten(AUSDRUCK)

fügt alle Teilausdrücke mit dem selben Operator wie AUSDRUCK unter einem

Operator zusammen. In unserem Beispiel wird dadurch aus einer Liste mit

Teillisten [A, [B, . . . , N]] eine gemeinsame Liste [A, B, . . . , N].

Die Funktion lastn(LISTE, n) liefert die letzten n Elemente der LISTE

zurück. In Kombination sorgen die beiden Funktionen in unserem Algorith-

mus dafür, dass die ersten beiden Elemente von Resten und Moduli jeweils

durch die berechnete Teillösung ersetzt und dann als neue Liste in den re-

kursiven Funktionsaufruf eingesetzt werden können.

Dass der Algorithmus korrekt arbeitet, zeigt der Vergleich mit Maxima’s ei-

gener Implementierung:

(%i44) [a,b] : [[35,24,17], [89,27,23]]$

(%i45) chinese remainder(a,b);

(%o45) 2616

(%i46) chinese(a,b);

(%o46) 2616

Falls der Vergleich mit Maxima’s Implementierung nicht genug ist, können

wir das Ergebnis auch für alle Gleichungen prüfen:

(%i47) map(lambda([i,j], is(mod(2616,j) = i)), a, b);

(%o47) [true,true,true]

Die obige Codezeile führt dabei zwei wichtige Konzepte der Programmie-

rung mit Maxima ein:

Die Funktion map(FUNKTION, AUSDRUCK 1, . . . , AUSDRUCK n) behält da-

bei die Struktur der Ausdrücke bei, wendet aber FUNKTION auf jeden ein-

zelnen AUSDRUCK an. Als Funktion haben wir einen sogenannten lambda-

Ausdruck angegeben. lambda-Ausdrücke fungieren als namenlose (anonyme)

Funktionen und simulieren demnach die Struktur einer Funktion mitsamt

Ein-/Ausgabe. Der lambda-Ausdruck in Zeile 47 simuliert damit eine Funk-

tion mit Eingabeparametern i und j, welche true/false zurückliefert, je

nachdem, ob das Ergebnis aus Zeile 45/46 modulo j gleich i ist. Der Aufruf

von map wendet diese Überprüfung dann auf die Listen a und b an, wobei i

die Liste (a) der Reste durchläuft und j die Liste (b) der Moduli.

Die Erklärung von Zeile 47 mag etwas ausgiebig erscheinen, ist aber essenti-

ell, denn lambda-Ausdrücke stellen ein Schlüsselkonzept in der funktionalen

Programmierung dar.

Abschließend bleibt zu unserer Implementierung noch zu sagen, dass diese

kein Error-Handling beinhaltet. Ein gutes Error-Handling ist in den meisten

Programmiersprachen/CAS oftmals genau so wichtig wie die Funktionalität

des Programms selbst. Besonders, wie in unserem Fall, rekursive Methoden,

laufen bei schlechtem Error-Handling oft Gefahr, unendlich tief (so tief wie

möglich) zu rekurrieren und somit potentiell nicht nur das Programm son-

dern den ganzen PC zum Absturz zu bringen7.

5.3 Modulare Quadratwurzeln

Auch in der modularen Arithmetik können Operationen wie beispielsweise

das Wurzelziehen definiert werden, auch wenn die Aussagen hier meistens

nicht so trivial ausfallen wie in ihrem nicht-modularen Äquivalent:

Definition 5.8. Sei N ∈ N mit N ≥ 2. Besitzt die Gleichung:

x2 ≡ a (modN)

eine Lösung, so nennen wir die Lösung x :=
√
a (modN) modulare Quadrat-

wurzel von a. Weiter nennen wir a in diesem Fall einen quadratischen

Rest modulo N .

Dass nicht jede Zahl eine modulare Quadratwurzel besitzt, zeigt folgende

Eingabe:

(%i48) map(lambda([x], mod(x^2,10)), makelist(n, n, 0, 9));

(%o48) [0,1,4,9,6,5,6,9,4,1]

Man sieht, dass nur die Zahlen 0,1,4,5,6,9 quadratische Reste in Z10 sind.

Die Zahl 7 dahingegen lässt sich nicht durch a2, a ∈ Z10 darstellen.

Die Funktion makelist erzeugt hier eine Liste der Zahlen von 0 bis 9.

7CAS wie Maxima und Mathematica setzen für ihre Berechnungen einen eigenen Kernel
auf und verhindern dadurch automatisch solche PC-übergreifenden Abstürze. Dennoch ist
beim iterativen/rekursiven programmieren Vorsicht geboten! Weiterhin bietet ein gutes
Error-Handling eine wichtige Schnittstelle zum externen Benutzer.

Dabei ist das erste n der Ausdruck für jedes Listenelement (wir wollen le-

diglich die Zahl an sich) und das zweite n der Name der Variablen, die die

Zahlen von 0 bis 9 durchläuft.

Aus Zeile 48 können wir uns somit auch einen relativ simplen Algorithmus

zur Bestimmung der modularen Quadratwurzel herleiten:

(%i49) mod root(x,p) := block([quad],

quad : map(lambda([a],mod(a^2,p)),makelist(n,n,0,p)),

sublist indices(quad,lambda([y],x = y)),

if %% = [] then (

printf(true,

concat("∼d besitzt keine modulare",

"Quadratwurzel modulo ∼d")
x,p),

return(-1)) else

return(map(lambda([y], y-1),%%)))$

(%i50) map(lambda([x], mod(x^2,13)), makelist(n, n, 0, 12));

(%o50) [0,1,4,9,3,12,10,10,12,3,9,4,1]

(%i51) mod root(11,13);

11 besitzt keine modulare Quadratwurzel modulo 13

(%o51) -1

(%i52) mod root(12,13);

(%o52) [5,8]

Unser Algorithmus berechnet zunächst Quadrate für alle a ∈ Zp. Die Funk-

tion sublist indices(LISTE, WAHRHEITSFUNKTION) liefert alle Indizes aus

LISTE zurück, für welche die WAHRHEITSFUNKTION true zurückgeliefert hat

(in unserem Fall der simple Vergleich mit x). Mit %% kann in Funktions-

blöcken das Ergebnis der letzten Anweisung aufgerufen werden.

Ist also die Liste der Indizes leer, so erfüllt kein a ∈ Zp die Bedingung und

x besitzt keine modulare Quadratwurzel in Zp. Dies wird mit printf (prin-

ting formatted) ausgegeben. In der printf-Methode werden die benötigten

Variablen und deren Typ mit ∼ direkt im Text
”
angekündigt“, sodass an-

hand des Typs bereits eine geeignete Formatierung gestaltet werden kann.

In unserem Beispiel erwartet printf anhand der beiden ∼d im Text zwei

Dezimalzahlen, x und p.

Die Methode concat fügt die beiden Teilsätze zu einem Satz zusammen8.

Bevor wir uns einem weiteren Problem der modularen Arithmetik zuwenden,

soll zunächst eine optimierte Methode zur Berechnung der modularen Qua-

dratwurzeln vorgestellt werden. Wie man direkt sieht, ist die obige Variante

für große Moduli deutlich zu aufwendig. Wir wollen nun einen effizienteren

Algorithmus angeben, denn Maxima bietet hier ohne Zusatzpakete keine na-

tive Implementierung. Betrachten wir zunächst eine wichtige Definition aus

der Zahlentheorie:

Definition 5.9. Sei p prim und a ∈ Z mit gcd(a, p) = 1. Dann ist das

Legendre-Symbol definiert durch:

(
a

p

)
=

+1, falls a quadratischer Rest mod p ist

−1, falls a kein quadratischer Rest mod p ist.

Bemerkung 5.10. Eine Erweiterung des Legendre-Symbols stellt das Jacobi-

Symbol
(
a
m

)
dar, welches auch für zusammengesetzte Zahlen m definiert

ist. Das Jacobi-Symbol ist aber im Falle m prim äquivalent zum Legendre-

Symbol. In den meisten CAS findet sich daher üblicherweise nur eine Funk-

tion zum Bestimmen des Jacobi-Symbols.

Da der Algorithmus sehr umfangreich ist, soll hier nur kurz die Idee dahin-

ter geschildert werden. Auf den nächsten Seiten wird der Algorithmus dann

in seinem vollen Umfang dargestellt.

Konstruktion: Um effizient die Quadratwurzeln r2 ≡ a (mod p), p prim

zu bestimmen, wird zunächst eine Fallunterscheidung für p getroffen:

1. Fall (p ≡ 3 (mod 4)): Dann gilt direkt r ≡ ±a p+1
4 (mod p).

2. Fall (p ≡ 5 (mod 8)): Dann ergibt sich durch v ≡ (2a)
p−5
8 (mod p) und

i ≡ 2av2 (mod p) die Lösung r ≡ ±av(i− 1) (mod p).

3. Fall (p ≡ 1 (mod 8)): Dann können wir den Algorithmus von Tonelli-

Shanks benutzen, um eine Lösung r zu bestimmen.

8concat kann beliebige Elemente zusammenfügen, nicht nur Strings. In unserem Bei-
spiel ist die Verwendung von concat eigentlich nicht nötig, da der Satz auch in eine Zeile
geschrieben werden kann. Um aber eine gute Anschaulichkeit und einen guten Codestil
zu bewahren, wird der Satz hier in 2 Teilsätze aufgetrennt. concat bietet die einfachste
Möglichkeit, zwei Sätze ohne unnötigen Zwischenraum zusammenzufügen.

5 MODULARE ARITHMETIK

Die formale Ablaufvorschrift für den Algorithmus aus Fall 3 wollen wir hier

nicht ausführen.

Die Logik hinter der obigen Konstruktionsvorschrift stammt überwiegend di-

rekt aus den Folgerungen und Sätzen rund um das Jacobi-Symbol.

(%i53) mod sqrt(a,p) := block([d,e:0,i,k,q,r,r1,r2,v,w,x,y],

if jacobi(a,p) # 1 then (

printf(true,

"∼d ist kein quadratischer Rest modulo ∼d",
a,p),

return(-1)

)

else if mod(p,4) = 3 then (

r1 : power mod(a,(p+1)/4, p),

r2 : mod(-r1, p),

return([r1,r2])

)

else if mod(p,8) = 5 then (

v : power mod(2·a, (p-5)/8, p),

i : mod(2·a·v^2, p),

r1 : mod(a·v·(i-1), p),

r2 : mod(-r1,p),

return([r1,r2])

)

else if mod(p,8) = 1 then (

q : p-1,

while integerp(q/2) do

[q,e] : [q/2,e+1],

i : random(p)^q,

while power mod(i, 2^(e-1), p) = 1 do

i : random(p)^q,

[y,r,x] : [i,e,power mod(a, (q-1)/2, p)],

v : mod(a·x,p),
w : mod(v·x,p),

Wird auf der nächsten Seite fortgesetzt...

while w # 1 do (

k : 1,

while power mod(w, 2^k, p) # 1 do

k : k + 1,

d : power mod(y,2^(r-k-1),p),

y : power mod(d,2,p),

[r,v,w] : [k,mod(d·v,p),mod(w·y,p)]
),

[r1,r2] : [v, mod(-v,p)],

return([r1,r2])

)

)$

Auch wenn der Algorithmus zunächst sehr komplex und aufwendig erscheint,

so kann man doch recht schnell verifizieren, wie effizient er arbeitet:

(%i54) p : next prime(10^1000)$

(%i55) jacobi(7,p);

(%o55) 1

(%i56) [a,b] : mod sqrt(7,p)$

(%i57) mod(a^2,p);

(%o57) 7

(%i58) mod(b^2, p);

(%o58) 7

Dies zeigt zusätzlich sehr deutlich, wie effizient die Funktion power mod die

Potenzierung in Zp durchführt. Die Anzahl der Ziffern von a und b beträgt

jeweils 1000.

Zu obigem Algorithmus bleibt noch anzumerken, dass immer zwei Lösungen

erwartet werden, falls eine Lösung existiert. Dies folgt aus der Tatsache, dass

eine Zahl n modulo einer Primzahl p entweder keine modulare Quadratwur-

zel besitzt, oder genau zwei.

Anhand der Methode zur Berechnung von modularen Quadratwurzeln in Zp
soll noch eine Methode für Zp2 vorgestellt werden, die in einem späteren

Abschnitt über das Faktorisieren von Zahlen relevant wird:

5 MODULARE ARITHMETIK

Satz 5.11. Sei p > 2 prim und a ein quadratischer Rest modulo p.

Betrachte folgende Konstruktionsvorschrift:

1. Bestimme x0 mit x20 ≡ a (mod p).

2. Berechne xi ≡ (2x0)
−1 (mod p2).

3. Berechne x = xi · (x20 + a).

4. Berechne r ≡ x (mod p2).

Dann gilt r2 ≡ a (mod p2), d.h. r ist modulare Quadratwurzel von

a modulo p2.

Die 4 genannten Schritte lassen sich unter Verwendung unserer mod sqrt-

Funktion sehr einfach in Maxima überführen:

(%i59) mod p2 sqrt(a,p) := block([x,x0,xi],

x0 : mod sqrt(a,p)[1],

xi : inv mod(2·x0, p^2),

x : xi·(x0^2 + a),

return(mod(x, p^2))

)$

(%i60) mod p2 sqrt(16,5);

(%o60) 21

(%i61) mod(21^2, 25);

(%o61) 16

5.4 Der diskrete Logarithmus

Wie wir gesehen haben, lassen sich viele algebraische (und auch analyti-

sche) Konzepte mit, wenn nötig, einigen Zusatzinformationen, in die modu-

lare Arithmetik überführen. So auch der Logarithmus:

Definition 5.12. Seien g, h ∈ F×p (:= Z×p , p prim) und sei L(h) die Lösung

folgender Gleichung:

gL(h) ≡ h (mod p)

Dann nennen wir L(h) den diskreten Logarithmus von h modulo p

bezüglich g.

Die oben definierte Menge F×p erspart uns dabei etwas Schreibarbeit. Wir

erinnern uns an die Besonderheit, dass Zp ein Körper ist, falls p prim ist.

Die Einheitengruppe F×p enthält demnach alle Elemente {1, . . . , p−1}, da in

einem Körper alle Elemente außer die 0 invertierbar und demnach Einheiten

sind. Dass diese Gleichung nicht immer eine Lösung besitzen muss, zeigt fol-

gendes einfaches Beispiel:

(%i62) makelist(mod(2^n, 7), n, 1, 10);

(%o62) [2,4,1,2,4,1,2,4,1,2]

Hier sehen wir die ersten zehn 2-er Potenzen in Z7. Offensichtlich zeichnet

sich hier ein 3-er Zyklus [2,4,1] ab, der sich mit wachsendem Expontenten

immer wiederholt. Dass tatsächlich eine Periodizität vorliegt, zeigt folgende

einfache Rechnung:

20 = 1; 21 = 2; 22 = 4; 23 = 1

n ∈ N : 2n = 2k·3 · 2l = (23)k · 2l = 1k · 2l = 2l mit l ∈ {0, 1, 2}
⇒ 2n ∈ {2, 4, 1}

Die oben bewiesene Eigenschaft der 2-er Potenzen impliziert damit aber auch

folgende Aussage:

Die Gleichung 2k ≡ 5 (mod 7) besitzt keine Lösung.

Damit also die Gleichung gL(h) ≡ h (mod p) aus Definiton 5.12 eine Lösung

besitzt, muss noch eine Zusatzbedingung an g gestellt werden:

Definition u. Satz 5.13. Sei a ∈ Z×n . Die Ordnung von a modulo n ist

die kleinste Zahl k > 0, sodass ak ≡ 1 (modn) ist (ord(a)).

Ist a ∈ F×p so teilt ord(a) stets p− 1.

Gilt für die Ordnung eines Elementes ord(a) = p − 1, so nennen wir a Pri-

mitivwurzel (Erzeuger) modulo p.

Ist ein a ∈ F×p Primitivwurzel, so sieht man schnell, dass die Potenzen von

aq, q ∈ {1, · · · , p−1} jedes Element aus F×p darstellen, da die kleinste Potenz

die wieder zu 1 wird, p− 1 ist (daher auch der Name Erzeuger).

Wissen wir also, dass g ein Erzeuger der Gruppe F×p ist, so können wir mit

Sicherheit sagen, dass für alle h ∈ F×p ein diskreter Logarithmus L(h) modulo

p bezüglich g existiert.

Maxima bietet eine eigene Funktion zur Berechnung des diskreten Logarith-

mus:

(%i63) p : 113$

(%i64) g : zn primroot(p);

(%o64) 3

(%i65) zn log(93,g,p);

(%o65) 51

(%i66) mod(3^51, 113);

(%o66) 93

Die Funktion zn primroot(p) berechnet die kleinste Primitivwurzel in F×p .

Der Aufruf von zn log(h,g,p) löst dann die Gleichung gk ≡ h (mod p).

Zeile 66 beweist, dass die Funktion korrekt gearbeitet hat.

Würde man einen Algorithmus zum Bestimmen des diskreten Logarithmus

implementieren, so könnte man ganz naiv alle Potenzen 1, . . . , p− 1 durchpro-

bieren, bis man die Lösung gefunden hat. Diese
”
Brute-Force“-Methode büßt

aber mit größer werdendem p sehr schnell an Effizienz ein. Im Abschnitt über

Kryptographie werden wir uns deshalb noch einmal mit einem effizienteren

Algorithmus zur Berechnung des diskreten Logarithmus beschäftigen.

6 Polynomarithmetik

Bevor wir uns im nächsten Abschnitt mit den direkten Folgen der modularen

Arithmetik für die Kryptographie befassen, wollen wir zunächst einen Blick

auf Maxima’s Funktionen zur Polynomarithmetik werfen. Die Arbeit mit Po-

lynomen stellt in vielen Teilgebieten der Algebra eine essentielle Komponente

dar. Wir erweitern hier die Definition unseres Rings R um eine Veränderliche

x und erhalten somit den Polynomring R[x] der Polynome über R mit der

Darstellung
∑n

i=0 aix
i, ai ∈ R, an 6= 0.

Wir wollen zu Beginn ein paar elementare Funktionen für Polynome kennenler-

nen:

(%i67) f[x] := x^3 + 3·x^2 + 4·x + 5$

(%i68) f[4];

(%o68) 133;

(%i69) g[x] := (x+3)^5;

(%o69) gx:=(x+3)^5

(%i70) expand(g[x]);

(%o70) x5 + 15x4 + 90x3 + 270x2 + 405x + 243

(%i71) t : x^2 - 3·x + 2$

(%i72) factor(t);

(%o72) (x-1)(x-2)

Mit p[x] := POLYNOM definiert man das Polynom als Funktion, sodass man

direkt Werte für x einsetzen kann. Wie man in der Eingabezeile 71 sieht,

kann Maxima aber auch mit Polynomen als einfache Zuweisung arbeiten.

Die Funktion expand(AUSDRUCK) versucht so weit wie möglich, AUSDRUCK

zu expandieren, in dem beispielsweise Produkte und Potenzen ausmultipli-

ziert werden. Die Funktion factor(AUSDRUCK) arbeitet genau gegenläufig

und versucht, AUSDRUCK in irreduzibele Elemente über den ganzen Zahlen

zu zerlegen. Eine weitere wichtige Funktion bietet coeff(P,TERM). Mit ihr

erhält man den zu TERM zugehörigen Koeffizienten:

(%i73) coeff(t,x);

(%o73) -3

Will man eine Liste aller Koeffizienten haben, so muss unter Angabe des

Grads des Polynoms eine eigene Funktion definieren:

(%i74) all coef(P,n) := block([],

map(lambda([t], coeff(P,x,t)), makelist(k,k,n,0,-1))

)$

(%i75) all coef(t,2);

(%o75) [1,-3,2]

Die Funktion coeff(P,x,t) liefert den Koeffizienten des Terms xt. Die−1 als

letztes Argument von makelist legt die Schrittweite bei der Erstellung der

Liste fest. Hierdurch werden die Zahlen von n bis 0 rückwärts durchlaufen.

Anhand der Definition des Polynomrings R[x] lassen sich auch Eigenschaften

von Z auf R[x] übertragen, insbesondere auch die Division mit Rest:

Satz 6.1. Seien K Körper und a(x), b(x) ∈ K[x]. Dann existiert genau ein

Paar (q(x), r(x)) ∈ K[x] mit:

a(x) = q(x)b(x) + r(x) und deg(r(x), x) < deg(b(x), x)

deg(p(x), x) beschreibt dabei den Grad von p(x) in x, d.h.:

Ist p(x) =
∑n

i=0 aix
i, so ist deg(p(x), x) = maxi{ai 6= 0}, also die größte

x-Potenz von p(x).

In Maxima erhält man q(x) und r(x) mit den Funktionen quotient resp.

remainder:

(%i76) p : x^6 + x^3 + 2$

(%i77) m : x^2 + 4·x + 3$

(%i78) quotient(p,m);

(%o78) x^4-4x^3+13x^2-39x+117

(%i79) remainder(p,m);

(%o79) -351x - 349

Die Division mit Rest impliziert auch einen größten gemeinsamen Teiler in

K[x] sowie die Existenz von Bézoutkoeffizienten. Da wir uns hier in K[x] be-

finden, kann es passieren, dass der größte gemeinsame Teiler das konstante

Polynom 1 ist. Dies ist der Fall, wenn die beiden Polynome keinen gemein-

samen Faktor besitzen oder eins der Polynome/beide Polynome nicht über

K[x] faktorisierbar ist/sind:

(%i80) ezgcd(p,m);

(%o80) [1,x^6+x^3+2,x^2+4x+3]

Die Funktion ezgcd bestimmt zunächst den größten gemeinsamen Teiler d

von p(x) und m(x) und teilt die Polynome dann durch d. Die Ergebnisse der

Division sind der zweite und dritte Listeneintrag der Ergebnisliste.

Diese kurze Abschnitt sollte eine kleine Einführung in Maxima’s Umgang

mit Polynomen geben. Natürlich existiert noch ein enormes Spektrum an

weiteren Funktionen, beispielsweise zum Vereinfachen von rationalen Poly-

nomen, für die Polynomarithmetik an sich und die Polynomarithmetik in

Restklassenringen, diese sollen aber hier nicht weiter behandelt werden.

7 Kryptographie

In diesem Abschnitt wollen wir uns mit den algebraischen Anwendungen

in der Kryptographie beschäftigen. Dabei sollen verschiedene Kryptosyste-

me vorgestellt werden und ein besonderer Fokus darauf gelegt werden, wie

wichtig alte und moderne Algorithmen der Computeralgebra sind, um die

Sicherheit dieser Systeme einschätzen zu können.

7.1 Einführung und Private-Key-Verfahren

Wir beginnen zunächst mit der Definition einiger wichtiger Begriffe:

Definition 7.1. Sei A eine nicht-leere, endliche Menge und

(a1, a2, . . . , an) ∈ An =: A× · · · × A︸ ︷︷ ︸
n-mal

ein n-Tupel aus dem n-fachen Produkt

von A. Dann nennen wir A ein Alphabet und (a1, a2, . . . , an) einen Text

der Länge n über A. Die Wortmenge A∗ sind alle Texte beliebiger Länge

über A, d.h. A∗ = ∪n∈NAn.

Der Einfachheit halber werden wir in Zukunft die MengeA = {A,B, . . . , Z}
das natürliche Alphabet nennen. Wenn wir mit dem natürlichen Alphabet

rechnen müssen, weisen wir jedem Buchstaben seinen zugehörigen Platz im

Alphabet zu, beginnend bei 0, d.h. A = {A,B, . . . , Z} = {0, 1, . . . , 25}.
Außerdem muss klar sein, was wir unter einer Verschlüsselung verstehen:

Definition 7.2. SeienA,B zwei Alphabete undA∗,B∗ die zugehörigen Wort-

mengen. Ein Private-Key-Verfahren ist ein Tupel (W , C,K, E,D) mit fol-

genden Definitionen:

• W ist die Menge der zulässigen Klartexte, d.h. die Menge der erlaub-

ten Texte, die verschlüsselt werden können (meistens P = A∗).

• C ist die Menge der Chiffretexte, d.h. die Menge der Texte, die durch

die Verschlüsselung entstehen können (meistens C = B∗).

• K ist der Schlüsselraum, d.h. die Menge aller zulässigen Schlüssel.

• E :W ×K → C, w 7→ E(p) := c heißt Verschlüsselungsfunktion.

• D : C × K → W , c 7→ D(c) := w heißt Entschlüsselungsfunktion.

Wird die Verschlüsselungsfunktion E (encrypt) auf einen Text w ∈ P ange-

wandt, nennen wir dies die Verschlüsselung von w.

Wird die Entschlüsselungsfunktion D (decrypt) auf einen Text c ∈ C ange-

wandt, nennen wir dies die Entschlüsselung von c.

Nachdem die wichtigsten Begriffe geklärt sind, wollen wir uns einigen histo-

rischen Beispielen samt Maxima-Implementierung widmen:

Die Vigenère-Chiffre: Seien A das natürliche Alphabet,W = C = A∗ und

s ∈ N. Wir nennen die bijektive Abbildung σk : A → A, a 7→ a+ k (mod 26)

Links-Shift (s ≥ 0) bzw. Rechts-Shift (s < 0). Als Schlüsselraum wählen

wir K = {0, . . . , 25}s. Für einen Schlüssel k = (k1, . . . , ks) und einen Text

w = w1 · · ·wn ∈ W der Länge n definieren wir:

E(k, w) = E(k, w1) · · ·E(k, wn)

= σk1(w1) · · ·σks(ws)σk1(ws+1) · · ·σks(w2s) · · ·
= σk(w) := c

und für c = c1 · · · cn ∈ C:

D(k, x) = D(k, c1) · · ·D(k, cn)

= σ−k1(c1) · · ·σ−ks(cs)σ−k1(cs+1) · · ·σ−ks(c2s) · · ·
= σ−k(c) := w

Wir ordnen also jedem Buchstaben aus w sein Chiffrat basierend auf dem

Schlüssel k zu. Überschreitet die Länge von w dabei die Länge von k, wird

der Schlüssel wieder von Vorne angewandt.

Folgendes Beispiel soll die Funktionsweise verdeutlichen (k = (1, 3, 5)):

Wort KRYPTOGRAPHIE

Zerlegung K R Y P T O G R A P H I E

+ Schlüssel 1 3 5 1 3 5 1 3 5 1 3 5 1

Chiffrat L U D Q W T H U F Q K N F

Um die Vigenère-Chiffre in Maxima zu implementieren benötigen wir zunächst

Funktionen zur Umwandlung Buchstabe � Zahl:

(%i81) text to num(s) := block([],

split(s,""),

map(lambda([c], cint(c) - 65), %%)

)$

(%i82) num to text(l) := block([],

map(lambda([i], ascii(i + 65)), l),

simplode(%%)

)$

(%i83) text to num("KRYPTOGRAPHIE");

(%o83) [10,17,24,15,19,14,6,17,0,15,7,8,4]

(%i84) num to text(%);

(%o84) KRYPTOGRAPHIE

Die Funktion split(s,"") trennt das Wort s in einzelne Buchstaben und

liefert das Ergebnis als Liste zurück9. cint(CHAR) liefert den zu CHAR gehöri-

gen ASCII-Code. Da die Großbuchstaben in dem ASCII-Code, den Maxima

verwendet, bei 65 beginnen, muss das Ergebnis noch verrechnet werden. An-

dersherum liefert die Funktion ascii(CODE) den zu CODE gehörigen Charak-

ter. Die Funktion simplode fügt die Liste aus Buchstaben wieder zu einem

Wort zusammen.

Dies führt uns zu folgendem Programm:

(%i85) encode vigenere(k,w) := block([],

w : text to num(w),

k : flatten(makelist(k,

ceiling(length(w)/length(k)))),

k : firstn(k, length(w)),

num to text(mod(w+k, 26))

)$

9Anstelle von "" hätte man hier auch z.B ein Leerzeichen " " eingeben können, dann
würde s nur bei allen Leerzeichen getrennt werden.

38



7 KRYPTOGRAPHIE

(%i86) decode vigenere(k,c) := block([],

c : text to num(c),

k : flatten(makelist(-k,

ceiling(length(c)/length(k)))),

k : firstn(k, length(c)),

num to text(mod(c+k, 26))

)$

(%i87) encode vigenere([1,3,5], "KRYPTOGRAPHIE");

(%o87) LUDQWTHUFQKNF

(%i88) decode vigenere([1,3,5], %);

(%o88) KRYPTOGRAPHIE

Die Caesar-Chiffre: Sie ist wohl eine der ältesten und bekanntesten Ver-

schlüsselungsarten. Sie ist eine Abwandlung der Vigenère-Chiffre mit einer

Schlüssellänge s = 1:

(%i89) encode vigenere([3], "CAESARCODE");

(%o89) FDHVDUFRGH

Dass die Caesar-Chiffre in der modernen Kryptographie unbrauchbar ist lässt

sich durch einen einfachen Aufruf schnell zeigen:

(%i90) map(lambda([i],

num to text(mod(text to num(%) - i, 26))),

makelist(n,n,0,25));

(%o90) [FDHVDUFRGH,ECGUCTEQFG,DBFTBSDPEF,CAESARCODE,

BZDRZQBNCD,AYCQYPAMBC,. . . ]

Einfaches durchprobieren der Möglichkeiten liefert uns direkt den unver-

schlüsselten Text.

Eine weitere Abwandlung der Vigenère-Chiffre liefert die Vernam-Chiffre. In

diesem Fall ist die Schlüssellänge s gleich der Textlänge n. Diese Art der Ver-

schlüsselungen nennt man monoalphabetische Substitutionschiffren.

7.2 Asymmetrische Verschlüsselungsverfahren

Auch wenn gute Private-Key-Verfahren eine der sichersten Varianten der Ver-

schlüsselung darstellen, sind sie in der Praxis meistens eher unpraktikabel,

da die Schlüssel
”
per Hand“ ausgetauscht werden müssen. Besonders im In-

ternet, wo die Kryptographie eine zentrale Rolle spielt, ist dies unmöglich.

Um trotzdem ein gutes Maß an Sicherheit gewährleisten zu können, exis-

tieren die Public-Key-Verfahren, welche öffentliche und private Schlüssel

miteinander verbinden.

Folgende Definition stellt eines der bekanntesten Public-Key-Verfahren vor:

Definition 7.3. (RSA-Verfahren) Eine Nachricht N ∈ N soll verschlüsselt

werden. Das RSA-Verfahren operiert nach folgendem Protokoll:

(Empfänger)

• Wähle zwei zufällige, genügend große Primzahlen p und q.

• Bestimme n = p · q.

• Bestimme ϕ = (p− 1) · (q − 1).

• Bestimme ein zufälliges e ∈ N mit e < n und gcd(e, ϕ) = 1.

• Das Paar (e, n) wird öffentlich gemacht. (öffentlicher Schlüssel)

• Die Zahl d = e−1 (modn) wird geheim gehalten. (privater Schlüssel)

• Die Zahlen p, q und ϕ werden aus Sicherheitsgründen gelöscht.

(Absender)

• Die Nachricht N wird mit C = E(e,n)(N) = mod(N e, n) verschlüsselt.

(Falls N ≥ n ist, wird die Nachricht vorher in geeignete Blöcke zerlegt.)

(Empfänger)

• Die Nachricht kann mit N = D(e,n)(C) = mod(Cd, n) entschlüsselt

werden.

Um das RSA-Verfahren in Maxima implementieren zu können, benötigen

wir zunächst einige Hilfsfunktionen für die Umwandlung der Texte in die

Zahl N :

(%i91) text to N(s) := block([len,N : 0],

s : map(lambda([c], cint(c)), split(s,"")),

len : length(s),

for i : 1 thru len do (

N : N + 1000^(i-1) · lastn(s,1),
s : firstn(s, length(s)-1)

),

N[1]

)$

(%i92) text to N("Kryptographie");

(%o92) 75114121112116111103114097112104105101

(%i93) N to text(N) := block([s : []],

while N # 0 do (

s : append(s, [mod(N, 1000)]),

N : (N - mod(N, 1000)) / 1000

),

map(lambda([i], ascii(i)), reverse(s)),

simplode(%%)

)$

(%i94) N to text(’’%o92);

(%o94) Kryptographie

Die Schleifenbedingung N # 0 ist dabei äquivalent zur Überprüfung N 6= 0.

Die Funktion reverse(LISTE) erzeugt eine umgekehrte Version von LISTE.

Nun benötigen wir noch die Ver-/Entschlüsselungsfunktion:

(%i95) encrypt rsa(s) := power mod(text to N(s), e, n)$

(%i96) decrypt rsa(C) := N to text(power mod(C, d, n))$

Die Funktion powermod(a,n,m) liefert eine effiziente Implementierung für

die Rechnung an (modm), dabei liefert der Aufruf powermod(a,-1,m) das

modulare Inverse von a modulo m. (Die Funktion wurde bereits einmal im

Kontext der modularen Quadratwurzeln verwendet)

Mit allen nötigen Hilfsfunktionen ausgestattet, können wir nun das RSA-

Modul implementieren:

(%i97) rsa module() := block([],

p : next prime(random(10^100)),

q : next prime(random(10^100)),

n : p · q,
e : phi : (p-1) · (q-1),
while gcd(e, phi) # 1 do

e : next prime(random(10^50)),

d : power mod(e, -1, phi),

kill(p, q, phi)

return("RSA-Module initialized")

)$

Die Funktion kill(VAR 1, ..., VAR n) löscht die Variablen

VAR 1, ..., VAR n aus der aktuellen Maxima-Sitzung und gibt deren Spei-

cher für zukünftige Rechnungen wieder frei10.

Mit unserem RSA-Modul können wir nun beginnen, Nachrichten zu ver-

schlüsseln:

(%i98) message : "Execute Order 66"$

(%i99) set random state(seed)$

(%i100) rsa module();

(%o100) RSA-Module initialized;

(%i101) [e,n];

(%o101) [60037114172868838252528941048419617783718312434069,

152992780035440866392276030606[139Ziffern]

614543240523012234142022119287]

(%i102) encrypt rsa(message);

(%o102) 425789046471770868093133371662[139 Ziffern]

928934764214484623960358504322

(%i103) decrypt rsa(%);

(%o103) Execute Order 66

Die Sicherheit des RSA-Verfahrens resultiert offensichtlich daraus, dass, wie

bereits besprochen, die Faktorisierung für genügend große Zahlen n mit zwei

10Als Variablen können auch selbst erstellte Funktionen zählen. kill kann prinzipiell
alles, was vom Nutzer definiert ist, löschen.

großen Primfaktoren beinahe unmöglich ist / eine absurde Laufzeit hat11.

Trotz des hohen Sicherheitsstandards des RSA-Verfahrens werden auch die

Algorithmen zur Faktorisierung immer besser und stellen somit mit steigen-

der Rechenleistung ein immer höheres Sicherheitsrisiko dar. Wir wollen uns

im Zuge dessen noch einmal mit der Faktorisierung beschäftigen:

7.2.1 Faktorisierung nach der Methode von Fermat

Wir haben bereits gezeigt, dass unser naiver Algorithmus der optimierten

Probedivision zur Bestimmung der Primfaktorzerlegung mit immer größer

werdenden Zahlen sehr schnell an seine Grenzen gerät. Diese Erkenntnis

kommt nicht besonders überraschend, wir müssen nur einen Blick auf die

Definition unseres RSA-Moduls werfen:

Die Zahl n im Modul ist bewusst so konstruiert, dass sie zwei große Prim-

faktoren p und q enthält. Um das n aus unserem RSA-Beispiel zu faktorisie-

ren, müsste der Algorithmus im Fall p < q die Zahl n durch alle Primzahlen

bis zur Größenordnung 10100 dividieren, um p zu finden. Dies ist selbst auf

den modernsten Computern eine, in einem akzeptablen Zeitraum, unlösbare

Aufgabe.

Auch wenn das Problem der Faktorisierung im Sinne der Komplexitätstheo-

rie möglicherweise nie einen
”
effizienten“ Algorithmus zur Lösung besitzen

wird, so existieren doch Verfahren, welche deutlich effizienter arbeiten als un-

ser Algorithmus, wie der Vergleich mit Maxima’s ifactors-Funktion gezeigt

hat12.

Viele dieser Verfahren bedienen sich geschickt den Erkenntnissen der Zahlen-

theorie, um die Algorithmen zu optimieren. Eines dieser Verfahren ist das

Quadratische Sieb, welches in den folgenden Seiten besprochen und imple-

mentiert werden soll.

Zunächst wenden wir uns aber einer etwas älteren Faktorisierungsmethode

von Fermat zu, welche den Grundstein für das Quadratische Sieb legt:

11Auch wenn der Algorithmus die Aufgabe der Primfaktorzerlegung theoretisch immer
lösen kann, so sind beispielsweise potentielle Laufzeiten von 100 Jahren+ eher ungefährlich
für moderne Sicherheitssysteme.

12Der Verweis auf die Komplexitätstheorie spielt hier darauf an, dass zwar bekannt ist,
dass das Problem der Faktorisierung in NP liegt, aber nicht, ob nicht doch ein Algorithmus
zur Lösung mit polynomieller Laufzeit existiert.

Voraussetzung: Wir haben uns in Abschnitt 5 bereits mit modularen Qua-

dratwurzeln befasst. Eine wichtige Erkenntnis war, dass diese, wenn sie über-

haupt existieren, nicht eindeutig sein müssen. Im Falle einer zusammenge-

setzten Zahl N kann es sogar bis zu 4 modulare Quadratwurzeln (±b1,±b2)
als Lösung der Gleichung b2 ≡ a (modN) geben. Wir nehmen an, dass

N = p·q für zwei Primzahlen p, q ist. Der chinesische Restsatz verrät uns, dass

für die Lösungen ±b1 ∈ Zp und ±b2 ∈ Zq gelten muss (oder umgekehrt)13.

Hat man nun Kenntnis über zwei Zahlen x, y mit x2 ≡ y2 ≡ a (modN) und

x 6≡ ±y (modN), kann man eine Zerlegung von N konstruieren, denn aus

x2 ≡ y2 (modN) folgt direkt (x + y)(x − y) ≡ 0 (modN). Da N nach Vor-

aussetzung keinen dieser Faktoren teilt, erhalten wir mit gcd(x+ y,N) einen

nicht-trivialen Teiler von N .

Die Voraussetzungen implizieren eine bereits optimierte Version von folgen-

dem Verfahren:

Definition 7.4. (Faktorisierungsverfahren von Fermat) Sei N ∈ N
ungerade und zusammengesetzt. Konstruiere eine Faktorisierung von N nach

folgender Vorschrift:

• Wähle x0 := d
√
Ne

• Bilde für x = x0, x0+1, x0+2, . . . nacheinander die Differenzen x2−N ,

bis eine Quadratzahl entsteht.

• Dann ist x2 −N = y2 und es folgt daraus die Zerlegung:

N = x2 − y2 = (x+ y)(x− y)

Wir wollen dieses Verfahren in Maxima implementieren:

(%i104) factor fermat(N) := block([x : ceiling(sqrt(N))],

while not integerp(sqrt(x^2 - N)) do

x : x+1,

return([x + (sqrt(x^2 - N)), x - (sqrt(x^2 - N))])

)$

13Der chinesische Restsatz impliziert eine Isomorphie ZN
∼= Zp1

× · · · ×Zpn
, falls N die

Primfaktorzerlegung N = p1 · · · pn besitzt.

Das Fermat-Verfahren ist für eine große Zahl N nur dann geeignet, wenn die

Primzahlen, aus denen sie sich zusammensetzt, sehr nah beieinander liegen.

Dann ist der Algorithmus aber erstaunlich effizient:

(%i105) [p,q] : [next prime(10^20),next prime(10^21)]$

(%i106) N : p · q;
(%o106) 100000000000000000050700000000000000004563

(%i107) showtime: true$

(%i108) ifactors(N);

Evaluation took 6.5220 seconds (6.5940 elapsed)

using 2761.449 MB.

(%o108) [[100000000000000000039,1],[1000000000000000000117,1]]

(%i109) factor fermat(N);

(Muss abgebrochen werden)

(%i110) [p,q] : [next prime(10^20),

next prime(next prime(10^20))]$

(%i111) N : p · q;
(%o111) 10000000000000000016800000000000000005031

(%i112) ifactors(N);

Evaluation took 18.7360 seconds (18.8600 elapsed)

using 8069.978 MB.

(%o112) [[100000000000000000039,1],[100000000000000000129,1]]

(%i113) factor fermat(N);

Evaluation took 1.4320 seconds (1.4490 elapsed)

using 296.997 MB.

(%o113) [100000000000000000129,100000000000000000039]

(%i114) showtime: false$

Die Erweiterung des Fermat’schen Faktorisierungsverfahrens betrachtet eine

Reihe von Kongruenzen u2i ≡ vi (modN). Bilden eine Auswahl v1 · · · vn := y2

ein Quadrat, so ergibt sich eine Kongruenz u1 · · ·un =: x2 ≡ y2 (modN).

Um die richtigen vk zu finden, muss man nur die Primfaktorzerlegung der

vk betrachten und darauf achten, dass die Exponenten der Primfaktoren im

Produkt immer gerade sind. Hat man die nötigen vk gefunden, kann man x

und y berechnen und mit gcd(x + y,N) einen nicht-trivialen Teiler von N

bestimmen. Diese Idee führt uns nun zum Quadratischen Sieb:

7.2.2 Faktorisierung mit dem Quadratischen Sieb

Um die Auswahl der vk für die kommenden Algorithmen etwas konkreter zu

gestalten, beginnen wir mit folgender Definition:

Definition 7.5. (Auswahl für Kongruenzen) Um genügend quadratische

Reste u2i ≡ vi (modN) zu erhalten, die sich leicht faktorisieren lassen, legen

wir den Kongruenzen eine Faktorbasis F zugrunde, mit

F := {−1, 2, p2, . . . , pn−1} und 2 < pi ≤ L ∈ N für i = 2, . . . , n− 1

Bei der Auswahl der Kongruenzen betrachten wir nur die vi, welche sich

vollständig in Elemente aus F zerlegen lassen.

Angenommen, wir haben ein vi =
∏

0≤ν<n p
αi,ν
ν mit pν ∈ F gefunden. Da

für die Produktbildung nur wichtig ist, ob die αi,ν gerade sind, bilden wir zu

jedem der αi,ν ein βi,ν := αi,ν (mod 2) ∈ F2 und speichern den Vektor:

~βi = (βi,0, . . . , βi,n−1) ∈ Fn
2 (1)

Bemerkung 7.6. Maximal n der βi können linear unabhängig sein. Spätes-

tens beim Fund von vn+1 muss demnach ~βi,n+1 von vorher berechneten ~βi,j

linear abhängig sein, d.h. es existieren Indizes i1, . . . , ir mit:

~βi,1 + · · ·+ ~βi,r = ~0 ∈ Fn
2

Daraus folgt, dass
∏r

k=1 vi,k =: y2 eine Quadratzahl ist.

Das Auswahlkriterium für die Kongruenzen liefert uns zwar eine gute

Möglichkeit zu entscheiden, welche vi’s für das Faktorisierungsverfahren güns-

tig sind, rollt aber auch gleichzeitig ein altes Problem wieder auf. Um zu ent-

scheiden, ob ein vi das Auswahlkriterium erfüllt, müssten wir uns wieder der

Probedivision bedienen, um zu sehen, ob vi ausschließlich in Primfaktoren

aus der Faktorbasis zerfällt. An dieser Stelle setzt das Quadratische Sieb ein:

Definition 7.7. (Quadratisches Sieb) Werden die quadratischen Reste

von einem quadratischen Polynom Q(x) erzeugt (Fermat: Q(x) = x2−N), so

ist Q(ξ) = η genau dann durch eine Primzahl p teilbar, falls Q(ξ) ≡ 0 (mod p)

ist. Diese Gleichung hat zwei Lösungen x1 und x2 modulo p, also sind alle

Q(x1 + kp) und Q(x2 + kp) mit k ∈ Z durch p teilbar.

Dies spart die Division aller anderen Werte Q(x) durch p.

Ein einzelnes Polynom zum Erzeugen der quadratischen Reste birgt dabei

auch gewisse Risiken. Betrachten wir Q(x) beispielsweise auf einem Intervall

|x −
√
N | ≤ M , so kann es passieren, dass zu wenige quadratische Reste

komplett in Elemente aus der Faktorbasis zerfallen. Verändert man eine der

3 Komponenten, so wird entweder die Rechnung deutlich aufwendiger, oder

die Wahrscheinlichkeit eines geeigneten quadratischen Restes noch geringer.

Dies führt uns zur Konstruktion eines optimierten Quadratischen Siebs:

Definition 7.8. (Multipolynomiales Quadratisches Sieb (MPQS))

Sei Q(x) = ax2 +2bx+c; a, b, c ∈ Z mit der Diskriminante ∆ := b2−ac = N .

Dann sind alle Werte:

aQ(x) = (ax+ b)2 − (b2 − ac) = (ax+ b)2 −N

quadratische Reste modulo N . Wählen wir a = q2 mit q prim und q - N , so

sind auch alle Werte Q(x) quadratische Reste. Dies liefert eine zusätzliche

Einschränkung für die Faktorbasis:

Gilt für eine Primzahl p > 2 dass p|Q(x), so folgt:

(ax+ b)2 ≡ N (mod p)⇒
(
N
p

)
= +1

d.h. N ist quadratischer Rest modulo p. Wir konstruieren die zusätzlichen

Elemente der Faktorbasis F also nur aus Primzahlen p > 2, welche die Be-

dingung
(
N
p

)
= +1 erfüllen.

Daraus können wir zunächst eine Funktion zum Erstellen der Faktorbasis

gewinnen:

(%i115) gen factorbase(N,len) := block([F : [-1,2],k,p : 2],

for k : 3 thru len do (

p : next prime(p),

while jacobi(N,p) #1 do

p : next prime(p),

F : append(F, [p])

),

return(F)

)$

(%i116) Fb : gen factorbase(400003, 10);

(%o116) [1,2,3,7,29,31,37,43,71,73]

Als Nächstes benötigen wir die modularen Quadratwurzeln von N modulo

den Elementen aus F . Hierzu greifen wir auf unsere effiziente Implementie-

rung aus Abschnitt 5.3 zurück:

(%i117) N roots F(N,F) := block([k,roots : [1,1]],

for k : 3 thru length(F) do

roots : append(roots, [mod sqrt(N,F[k])[1]]),

return(roots)

)$

(%i118) N roots F(400003, Fb);

(%o118) [1,1,1,4,8,14,25,24,29,67]

Konstruktion der Polynome: Um die Polynome zu konstruieren, müssen

wir die Gleichung b2 − ac = N der Diskriminante lösen, dabei folgt daraus

b2 ≡ N (mod a). Wählen wir wie definiert a = q2 mit p > 2 prim, so muss

b2 ≡ N (mod q2) gelten, N muss also quadratischer Rest modulo q2 sein.

Wählen wir die Lösung 0 ≤ b < a = q2, so folgt aus b2 ≡ N (mod a), dass

a|b2 − N gilt und wir erhalten mit c := b2−N
a

eine Lösung der Ausgangs-

gleichung.

(%i119) make pol(N,q) := block([a,b,c,qq,tmp],

if is(primep(q)) then

qq : q else

qq : next prime(q),

if is(qq = 2) then

qq : next prime(qq),

while jacobi(N,qq) # 1 do

qq : next prime(qq),

a : qq^2,

b : mod p2 sqrt(N,qq),

c : floor((b^2-N) / a),

return([qq,a,b,c])

)$

(%i120) make pol(400003, 4);

(%o120) [7,49,4,-8163]

(%i121) %[3]^2 - %[2]·%[4]
(%o121) 400003

Siebtechnik: Zunächst müssen wir ein Intervall wählen, über dem wir die-

jenigen xi herausssieben, für die Q(xi) komplett in Elemente aus der Faktor-

basis zerfällt.

Ist x0 das Minimum von Q(x), so wollen wir die Schranke für das Inter-

vall |x − x0| ≤ M so wählen, dass M ≈ |Q(x0)|. Betrachten wir zunächst

die Gleichung aQ(x) = (ax + b)2 − N . Ist ein 2 < p ∈ F , so gilt p|Q(x)

genau dann, wenn (ax + b)2 ≡ N (mod p) ⇔ ax + b ≡ ±r (mod p) ist, wo-

bei r die modulare Quadratwurzel von N modulo p ist. Seien nun ξ1, ξ2 mit

ξ1,2 ≡ a−1(±r − b) (mod p), so ist Q(x) für x = ξ1 + kp und x = ξ2 + kp

mit k ∈ Z durch p teilbar. Anstatt nun für alle x den Wert Q(x) durch p

zu teilen, verfahren wir wie folgt: Nach der Initialisierung des Siebintervalls

addiere für jedes p ∈ F an den Stellen x = ξ
(p)
ν + kp den Wert log(p). Ist

Q(x) vollständig in Elemente aus F zerlegbar, so addieren sich die Logarith-

men zu log(|Q(x)|). Wählen wir aufgrund der Rechenungenauigkeit nun eine

Toleranzgrenze

t > 0, so können wir als potentielle Werte Q(x) diejenigen x in Betracht

ziehen, für welche die Summe der Logarithmen den Wert log(|Q(x)|) bis auf

die Toleranz t erreicht. Da sich der Wert log(|Q(x)|) nur sehr langsam mit

x ändert, können wir ihn auch auf dem gesamten Siebintervall durch sein

Maximum ersetzen, was zusätzliche Rechnungen einspart.

Bilden des Logarithmusvektors: Die Logarithmen müssen nicht exakt

sein, daher können wir die Rundung von log(p) · scale auf eine Ganzzahl

betrachten, wobei scale ein Skalierungsfaktor ist, hier scale = 128. Da der

Logarithmusvektor immer wieder benötigt wird und nicht ständig neu berech-

net werden soll, benutzen wir zur Implementierung Maxima’s Remember-

programmierung:

(%i122) scale : 128$

(%i123) log F[F] := block([k,lg F : [0]],

for k : 2 thru length(F) do

lg F : append(lg F,

[round(ev(log(F[k]),numer)·scale)]),
return(lg F)

)$

Wie im Beispiel für die rekursive Berechnung der Fibonacci-Zahlen verwen-

den wir eckige Klammern log F[F] anstatt runder Klammern log F(F).

Wird diese Funktion nun einmal auf eine Faktorbasis F angewandt, so wird

das Ergebnis des Aufrufs log F[F] gespeichert und beim zweiten Aufruf von

log F[F] abgerufen anstatt nochmal neu berechnet. Der Aufruf

ev(log(...),numer) sorgt dafür, dass der Logarithmus als Gleitkommazahl

dargestellt wird und somit gerundet werden kann.

Algorithmus: Wir wollen uns nun, nach der geleisteten Vorarbeit, dem

eigentlichen Algorithmus zuwenden. Da dieser recht komplex ist und sich über

mehrere Funktionen erstreckt, wird seine Beschreibung in Unterabschnitte

eingeteilt, in denen die Funktionsweise der jeweiligen Funktionen detailliert

beschrieben wird.

1. Initialisierung der Variablen: Zunächst werden alle Variablen definiert,

die von mehreren Funktionen des Algorithmus benutzt werden müssen. Wie

beim RSA-Modul werden diese Variablen global definiert und lassen sich so-

mit ohne Einschränkung im aktuellen Maxima-Kernel abrufen:

(%i124) QS init(N) := block([blen,i],

Num : N,

blen : bit length(N),

Flen : max(8, floor(blen^2 / 32)),

Srange : blen·256,
Startq : isqrt(floor(isqrt(2·N) / Srange)),

Fbas : gen factorbase(N,Flen),

Rvec : N roots F(N,Fbas),

Lvec : log F[Fbas],

Mat : ematrix(Flen+1,2·Flen+1,0,1,1),
Pivot : [],

for i : 1 thru Flen do

Pivot : append(Pivot, [i]),

CurRow : 1,

Count1 : 0, Count2 : 0,

Stack : [],

Sieve : makelist(0,n,2·Srange),
Vvec : makelist(0,n,Flen+1),

Uvec : makelist(0,n,Flen+1),

return(Flen)

)$

Da diese Variablen immer wieder in den kommenden Funktionen auftauchen,

sollen sie hier im Detail erklärt werden:

Num: Die zu faktorisierende Zahl N .

blen: Die Bit-Länge der Zahl N .

(Die Funktion bit length(n) liefert die Anzahl der Bits zurück, die nötig

sind, um n im Binärsystem zu repräsentieren (Bit-Länge).)

Flen: Die vorgegebene Anzahl der Elemente, die in der Faktorbasis enthal-

ten sein sollen.

Srange: Die Länge der Hälfte des Siebintervalls (das M aus der Erklärung

der Siebtechnik).

Startq: Der initiale Wert für die Funktion make pol(N,q), d.h. bei Startq

beginnt die Funktion in ihrem ersten Durchlauf die Suche nach einer geeigne-

ten Primzahl zum Erzeugen der Siebpolynome.

(Die Funktion isqrt(n) liefert den ganzzahligen Anteil von
√
n.)

Fbas: Die Faktorbasis zur Zahl N .

Rvec: Die modularen Quadratwurzeln von N bzgl. der Faktorbasis Fbas.

Lvec: Der skalierte Logarithmusvektor bzgl. der Faktorbasis Fbas.

Mat: Die Matrix die verwendet wird, um die ~βi aus Gleichung (1) aus De-

finition 7.5. auf lineare Unabhängigkeit zu testen. Die Logik dahinter ist

wie folgt: Angenommen die Vektoren a1, . . . , am ∈ Fn
2 sollen auf lineare Un-

abhängigkeit gestestet werden. Wir bilden zunächst die m× n-Matrix A mit

den Zeilen ai und erweitern diese mit der m-reihigen Einheitsmatrix Em zu

der m× (n+m)-Matrix B. Durch elementare Zeilenumformungen, die simul-

tan auf A und Em angewandt werden, bringen wir A auf Zeilenstufen-Form

und erhalten somit A′ aus A und C aus Em. Ist nun die k-te Zeile von A′

gleich null, so gilt für die k-te Zeile ck der Matrix C außerdem ckA = 0.

Daraus folgt: Die Summe der Zeilen j von A, für die gilt ckj = 1, ist 0.

(Die Funktion ematrix(m,n,x,i,j) erzeugt eine m × n-Matrix gefüllt mit

0-en, wobei das Element in Zeile i, Spalte j, den Wert x bekommt.)

Pivot: Diese Liste enthält die Zahlen von 1 bis Flen und wird für das oben

beschriebene Gauß’sche Eliminationsverfahren benötigt.

CurRow: Enthält den Index der Zeile von Mat, auf der die aktuellen Opera-

tionen stattfinden.

Count1: Zählt, wie viele verschiedene quadratische Polynome zum Erzeugen

der Kongruenzen benötigt werden.

Count2: Zählt, wie viele der gefunden quadratischen Reste komplett in

Primfaktoren aus der Faktorbasis Fbas zerfallen.

Stack: Speichert vorübergehend die Paare (u, v), die durch das Sieb-Verfahren

entstehen. Hier wird der abstrakte Datentyp Stack14 durch eine Liste imple-

mentiert.

Sieve: Eine Liste, die das Siebintervall repräsentiert.

Uvec, Vvec: Listen, die jeweils die Werte der Paare (u, v) speichern, welche

der Kongruenz u2 ≡ v (modN) genügen.

Damit sind alle nötigen globalen Variablen initialisiert. Nun können die ein-

zelnen Funktion besprochen werden, die später in der Hauptfunktion zu ei-

14In der Programmierung stellt ein abstrakter Datentyp (ADT) eine Struktur zur Daten-
speicherung mit Ein-/Ausgabefunktion dar. Dadurch dass ein ADT nur eine Beschreibung
der Funktionalität liefert, muss er in den Programmiersprachen separat implementiert
werden.

nem Verfahren zusammengefügt werden.

2. Das Sieb-Verfahren:

(%i125) get qres(N) := block([Q,UV],

while is(length(Stack) = 0) do (

Q : make pol(N,Startq), Count1 : Count1 + 1,

disp(concat("Polynom: ", Count1)),

do sieve(Q),

sieve results(Q),

Startq : Q[1] + 2

),

UV : pop(Stack),

return(UV)

)$

Die Funktion get qres(N) ermittelt zunächst, ob noch Paare (u, v) im Stack

vorhanden sind. Wenn ja, wird durch die Methode pop(Stack) das erste Ele-

ment aus Stack ausgelesen, gelöscht und zurückgegeben. Wenn nicht, wird

zunächst ein neues Polynom zum Erzeugen der Paare gebildet. Die Funktion

disp(...) gibt den Inhalt ... auf dem Bildschirm wieder. Der Unterschied

zur print-Methode besteht darin, dass disp den Inhalt während der Berech-

nung ausgibt und nicht erst, wenn die Berechnung durchgelaufen ist. Hier

wird damit die aktuelle Anzahl an erzeugten Polynomen ausgegeben.

Danach werden die Funktionen do sieve(Q) und sieve results(Q) mit

dem gerade erzeugten Polynom aufgerufen, welche das eigentliche Sieb-Verfah-

ren nach der zu Beginn besprochenen Siebtechnik durchführen.

(%i126) do sieve(Q) := block([a1,b1,i,i0,k,p,r,r1,s,z],

Sieve : makelist(0,n,2·Srange),
for k : 3 thru Flen do (

[p,z,r] : [Fbas[k],Lvec[k],Rvec[k]],

s : mod(-Srange,p),

if mod(Q[2],p) # 0 then (

a1 : inv mod(Q[2],p),

b1 : mod(Q[3],p),

r1 : mod((r-b1)·a1,p),
if is(r1 >= s) then

i0 : r1-s

else

i0 : p+r1-s,

for i : (i0+1) thru 2·Srange step p do

Sieve[i] : Sieve[i] + z,

r2 : mod((p-r-b1)·a1,p),
if is(r2 >= s) then

i0 : r2-s

else

i0 : p+r2-s,

for i : (i0+1) thru 2·Srange step p do

Sieve[i] : Sieve[i] + z

)

)

)$

Die Funktion do sieve initialisiert zunächst das Siebintervall mit 0-en. Da-

nach übernimmt sie den Teil der Siebtechnik, in dem für jedes p ∈ F an den

Stellen x = ξ
(p)
ν +kp mit k ∈ Z des Siebintervalls, der Wert des Logarithmus-

vektors der Primzahl aus der Faktorbasis addiert wird, wobei ξ
(p)
ν wie oben

beschrieben durch ξ
(p)
ν ≡ a−1(±r − b) (mod p) definiert ist.

Den zweiten Teil der Siebtechnik übernimmt dann die Funktion sieve results:

(%i127) sieve results(Q) := block([k,qinv,target,u,v,x]

target : round(ev(log(Num/Q[2])·scale,numer)
-Lvec[Flen]),

qinv : inv mod(Q[1],Num),

for k : 1 thru 2·Srange do (

if is(Sieve[k] >= target) then (

x : k-1-Srange,

u : Q[2]·x + Q[3],

v : (u + Q[3])·x + Q[4],

u : mod(qinv·u,Num),
Stack : push([u,v],Stack)

)

),

return(length(Stack))

)$

Die Funktion legt zunächst die Toleranzgrenze target fest und läuft dann

durch das von do sieve befüllte Siebintervall. Dabei werden nur diejenigen

Stellen berücksichtigt, die in der Summe über ihre logarithmierten Werte an

die Toleranzgrenze herangekommen sind. Aus diesen Werten werden dann die

Paare (u, v) konstruiert und im Stack abgelegt. Die Funktion push([u,v],

Stack) fügt dabei das Paar (u, v) als Liste an vorderster Stelle im Stack ein

und gibt diesen zurück.

3. Probedivision potentieller Werte für v: Nachdem wir die Anzahl der

potentiellen Paare (u, v) deutlich eingeschränkt haben, können wir nun für

die verbleibenden Kandidaten eine Probedivision durchführen, um zu sehen,

ob v wirklich vollständig in Elemente aus der Faktorbasis zerfällt:

(%i128) qr trialdiv(v) := block([i,found : false,m,p],

Mat[CurRow] : makelist(0,n,length(Mat[CurRow])),

if is(v < 0) then (

v : -v,

Mat[CurRow][1] : 1

),

Wird auf der nächsten Seite fortgesetzt...

for i : 2 thru Flen do (

p : Fbas[i],

m : 0,

while is(v > 1) and is(mod(v,p) = 0) do

[v,m] : [floor(v/p),m+1],

if is(oddp(m)) then

Mat[CurRow][i] : 1,

if is(v <= 1) then (

found : true,

Mat[CurRow][Flen+CurRow] : 1,

Count2 : Count2 + 1,

return(true)

)

),

if found then

return(true)

else

return(false)

)$

Die Funktion qr trialdiv(v) führt besagte Probedivision für v durch. Da-

bei wird zunächst die Zeile der Matrix Mat, auf der wir aktuell operieren, mit

0-en initialisiert. Danach agiert die Methode wie folgt:

1. Ist v negativ, so wird v durch −v ersetzt. Da demnach −1 mit einem

ungeraden Exponenten in der Primfaktorzerlegung von v auftaucht,

setzte den ersten Eintrag in der aktuellen Zeile der Matrix Mat auf 1.

(Errinerung: -1 ist immer der erste Eintrag in der Faktorbasis.)

2. Teile nun v sukzessive durch die Elemente der Faktorbasis. Dabei wird

so oft durch ein Element geteilt, bis das Ergebnis nicht mehr ganzzahlig

ist. Kann durch ein Element eine ungerade Anzahl oft geteilt werden,

setzte den zugehörigen Eintrag in der aktuellen Zeile der Matrix Mat

auf 1.

3. Erreicht v während dieses Verfahrens die Bedingung v ≤ 1, so lässt sich

v komplett in Elemente der Faktorbasis zerlegen. In diesem Fall wird

zusätzlich in der aktuellen Zeile der Matrix Mat das zugehörige Diago-

nalelement der Einheitsmatrix auf der rechten Seite auf 1 gesetzt.

Außerdem wird das Verfahren abgebrochen und true zurückgegeben.

4. Erreicht v diese Bedingung nicht, so lässt sich v nicht komplett in Ele-

mente aus der Faktorbasis zerlegen und es wird false zurückgegeben.

4. Test auf lineare Abhängigkeit: Hat ein v die Probedivision
”
bestanden“,

so wird überprüft, ob der neu gewonnene Exponentenvektor der Zerlegung

von v den Rang der Matrix Mat um 1 erweitert, oder durch eine Linearkombi-

nation der vorherigen Vektoren dargestellt werden kann.

(%i129) gauss elim(k) := block([bool : true,i,j,j0],

for i : 1 thru k-1 do (

if is(Mat[k][Pivot[i]] = 1) then

Mat[k] : mod(Mat[k] + Mat[i],2)

),

for i : k thru length(Pivot) do (

j : Pivot[i],

if is(Mat[k][j] = 1) then (

j0 : Pivot[k], Pivot[k] : j, Pivot[i] : j0,

bool : false,

return(0)

)

),

if bool then

return(true)

else

return(false)

)$

Die Funktion gauss elim(k) geht davon aus, dass die Matrix Mat bereits bis

zur (k − 1)-ten Zeile auf Zeilenstufen-Form gebracht wurde und setzt diese

Transformation bis zur k-ten Zeile fort. Hier wird zum ersten Mal die Va-

riable Pivot benutzt. Die Interaktion zwischen Mat und Pivot ist dabei wie

folgt:

Der Pivot-Vektor ist eine Permutation der Zahlen von 1 bis Flen. Dabei gilt

für alle i < k, dass die Komponente der i-ten Zeile der Matrix Mat mit dem

Index Pivot[i] gleich 1 ist. Die Komponenten der i-ten Zeile der Matrix

Mat mit Index Pivot[j] sind dabei für alle j < i gleich 0.

Für die Transformation werden zunächst alle Spalten der k-ten Zeile der Ma-

trix Mat mit einem Index Pivot[i] für i < k, die nicht 0 sind, durch Addition

vorheriger Zeilen zu 0 gemacht (Erste for-Schleife). Danach wird ein Index

j := Pivot[k] bestimmt, sodass die k-te Zeile in der j-ten Spalte den Eintrag

1 hat. Wird ein solcher Index gefunden, wird der Pivot-Vektor so angepasst,

dass die Funktion im nächsten Durchlauf die Transformation fortsetzen kann.

Kann dieser Index allerdings nicht gefunden werden, so wurde die k-te Zeile

durch die vorangegangen Zeilenumformungen zu 0 gemacht, es liegt also eine

lineare Abhängigkeit vor. In diesem Fall wird true zurückgegeben, andern-

falls false.

5. Finden der Abhängigkeiten: Liefert die Funktion gauss elim ein posi-

tives Ergebnis, so muss aus der Matrix Mat ausgelesen werden, welche Zeilen

addiert wurden, um die Nullzeile zu erzeugen. Per Konstruktion und Ver-

arbeitung der Matrix Mat sind das nun trivialerweise die Indizes Flen+i,

0 ≤ i ≤ k, der k-ten Zeile von Mat, welche den Wert 1 haben.

Dieses Verfahren übernimmt die Funktion get rel(row):

(%i130) get rel(row) := block([i, rel : []],

for i : 1 thru row do (

if is(Mat[row][Flen+i] = 1) then

rel : append(rel, [i])),

return(rel)

)$

6. Finden eines nicht-trivialen Teilers: Der aus Punkt 5. gefundene Re-

lationsvektor kann nun verwendet werden, um den Versuch zu starten, einen

nicht-trivialen Faktor von N zu finden. Die Funktion find factor arbeitet

dabei auf den Vektoren Uvec und Vvec nach der, am Ende des Fermat’schen

Faktorisierungsverfahrens, besprochenen Methode:

Seien k1, . . . , kr die Einträge des Relationsvektors, sowie ui := Uvec[ki] und

vi := Vvec[ki]. Dann gilt u2i ≡ vi (modN) für alle 0 ≤ i ≤ r. Weiterhin ist

z :=
∏r

i=1 vi =: y2 eine Quadratzahl. Setzten wir nun x :=
∏r

i=1 ui, so gilt:

x2 ≡ y2 (modN), d.h. (x+ y)(x− y) ≡ 0 (modN). Ein möglicher Teiler von

N kann also gesucht werden, in dem gcd(x+ y,N) berechnet wird.

(%i131) find factor(N,relvec) := block([d,i,k,u,v,v1,x,y],

k : relvec[1],

[x,v,y] : [Uvec[k],Vvec[k],1],

for i : 2 thru length(relvec) do (

k : relvec[i],

[u,v1] : [Uvec[k],Vvec[k]],

x : mod(x·u,N),
d : gcd(v,v1),

v : floor(v/d)·floor(v1/d),
y : mod(y·d,N)

),

y : mod(y·isqrt(v),N),
d : gcd(x+y,N),

if is(d <= 1) or is(d = N) then

d : 0,

return(d)

)$

Die Zwischenschritte der Funktion find factor(N,relvec) resultieren da-

bei aus folgender Überlegung:

Wir können zwar x modulo N berechnen, jedoch nicht y2 modulo N , da aus

y2 ≡ ŷ2 (modN) nicht y ≡ ŷ (modN) folgt. Außerdem wird das Produkt

über die vi sehr schnell sehr groß. Um diesem Problem zu entgehen, agiert

die Funktion nach folgender Logik:

Das Gesamtprodukt ist zwar ein Quadrat, die einzelnen Faktoren jedoch

nicht. Man kann also bei der sukzessive Produktbildung erwarten, dass nicht-

triviale gemeinsame Teiler der Faktoren auftreten. Betrachten wir beispiels-

weise das Produkt zweier Zahlen A und B mit d = gcd(A,B), also A = ad

und B = bd, so gilt AB = ab · d2. Man dividiert also die gemeinsamen Teiler

jeweils heraus und baut somit bereits ein Teil von y auf.

7. Zusammenführen der Funktionen: Zum Schluss führen wir alle bespro-

chenen Funktionen in einer Hauptfunktion zusammen, welche uns im Idealfall

einen nicht-trivialen Teiler der Zahl N liefert:

(%i132) QS factorize(N) := block([d,relvec,result : 0,u,v],

QS init(N),

disp(concat("Quadratisches Sieb der Länge: ",

2·Srange)),
disp(concat("Faktorbasis 2 ...", Fbas[Flen],

” der Länge ", Flen-1)),

nbfail : 0,

while is(nbfail < 32) do (

[u,v] : get qres(N),

if qr trialdiv(v) then (

Uvec[CurRow] : u,

Vvec[CurRow] : v,

if not gauss elim(CurRow) then (

CurRow : CurRow + 1

)

else (

relvec : get rel(CurRow),

disp("Relation gefunden: Suche Faktor"),

d : find factor(N,relvec),

if is(d > 0) then (

result : d,

return(0)

)

else

nbfail : nbfail + 1

)

)

),

printf(true,

concat("Resultat: ∼d, ∼%",
"Polynome: ∼d, ∼%",
"Komplett faktorisierte Reste: ∼d"),

result, Count1, Count2),

Wird auf der nächsten Seite fortgesetzt...

if is(result > 0) then

return("Faktor gefunden")

else

return("Kein Faktor gefunden")

)$

Die Funktion QS factorize(N) arbeitet nun alle besprochen Schritte der

Reihe nach ab und verarbeitet in einer Schleife dabei jeweils die positiven/

negativen (true/false) Rückgabewerte der einzelnen Methoden. Desweite-

ren wird eine Schranke nbfail angeführt. Wird mehr als 32 Mal ein Relationsvek-

tor gefunden, aus dem kein nicht-trivialer Faktor von N resultiert, so bricht

die Funktion ab. Außerdem findet sich als Neuerung in der printf-Methode

die Anweisung ∼%, welche einen Zeilenumbruch darstellt.

Ein einzelner Aufruf der Funktion setzt nun den Algorithmus in Gang:

(%i133) QS factorize(2^50+1);

Quadratisches Sieb der Länge: 26112

Faktorbasis 2 ... 1069 der Länge 80

Polynom: 1

Polynom: 2

Relation gefunden: Suche Faktor

Relation gefunden: Suche Faktor

Resultat: 27118601,

Polynome: 2,

Komplett faktorisierte Reste: 71

(%o133) Faktor gefunden

Zum Schluss bleibt noch anzumerken, dass unser Algorithmus deutlich we-

niger optimiert ist als die Vorlage von Otto Forster aus seinem Buch Algo-

rithmische Zahlentheorie, auf dem diese Implementierung basiert. Überprüft

man die Laufzeiten der einzelnen Funktionen, so fallen besonders die länge-

ren Laufzeiten der Funktionen do sieve, sieve results und qr trialdiv

ins Gewicht. Dies liegt vermutlich einerseits daran, dass Forster in seiner

Implementierung die sehr effizienten Bit-Operationen verwendet, anderer-

seits eventuell aber auch daran, dass die von Forster entwickelte Program-

miersprache Aribas in C programmiert ist und Maxima in Common Lisp.

Eine weitere Möglichkeit hierfür liegt in der Diskrepanz zwischen den Stack-

Implementierungen.

7.3 Kryptographie auf elliptischen Kurven

In diesem Abschnitt wollen wir eine kurze Einführung in die Kryptographie

auf elliptischen Kurven geben, um anschließend nochmals einen Blick auf

das diskrete Logarithmus-Problem zu werfen. Da das Gebiet um elliptische

Kurven ein sehr umfangreiches Teilgebiet der Algebra darstellt, sollen hier

nur die relevantesten Informationen geliefert werden.

Definition 7.9. Sei y2 = x3 + Ax + b eine Gleichung über einer Menge K,

nennen wir den Graph E : y2 = x3 + Ax+ b eine elliptische Kurve. Ist K

ein Körper, so nennen wir E(K) eine elliptische Kurve über K.

Wir werden elliptische Kurven in diesem Abschnitt immer über dem end-

lichen Körper Fp = Z×p betrachten, den wir in der modularen Arithmetik

eingeführt haben. Betrachten wir als Beispiel den Graphen der Gleichung

y2 = x3 − x:

(%i134) load(implicit plot)$

(%i135) implicit plot(y^2 = x^3 - x, [x,-5,5], [y,-5,5])$

Die Funktion load(implicit plot) lädt das Modul implicit plot in den ak-

tuellen Maxima-Kernel, welches benötigt wird, um nicht-lineare Gleichungen

zu zeichnen. Der Aufruf liefert folgende Grafik:

Die Definition einer elliptischen Kurve liefert uns folgende Konstruktion für

Addition und Subtraktion:

Definition u. Satz 7.10. Sei E : y2 +Ax+B eine elliptische Kurve. Ferner

seien P1 = (x1, y1) und P2 = (x2, y2) zwei Punkte auf E mit P1, P2 6= ∞.

Definiere P1 + P2 = P3 = (x3, y3) wie folgt:

• Falls x1 6= x2, dann ist:

x3 = m2 − x1 − x2 und y3 = m(x1 − x3)− y1 mit m = y2−y1
x2−x1 .

• Falls x1 = x2 aber y1 6= y2, dann ist P1 + P2 =∞.

• Falls P1 = P2 und y1 6= 0, dann ist:

x3 = m2 − 2x1 und y3 = m(x1 − x3)− y1 mit m =
3x21+A

2y1

• Falls P1 = P2 und y1 = 0, dann ist P1 + P2 =∞

Ferner sei P +∞ = P für alle P auf E.

(E,+) ist eine abelsche Gruppe mit ∞ als neutralem Element.

Die obige Konstruktion resultiert dabei aus dem Schnittpunkt zwischen

der Gerade durch P1 und P2 und der elliptischen Kurve. Die skalare Mul-

tiplikation kP ist dann definiert durch die k-fache Addition des Punktes

mit sich selber. Weiterhin gilt für die Subtraktion Q − P = Q + (−P ) mit

−P = (xP ,−yP ).

Die Definition einer elliptischen Kurve zzgl. einer Addition auf einer solchen

liefert uns folgende kryptographische Anwendung:

Definition 7.11. (Diffie-Hellman-Schlüsselaustausch auf

elliptischen Kurven) Angenommen zwei Personen (A,B) benutzen ein

Private-Key-Verfahren, um Nachrichten zu verschlüsseln. Im folgenden Proto-

koll soll ein Algorithmus definiert werden, um den Schlüssel des Verfahrens

in einem öffentlichen Kommunikationsraum sicher austauschen zu können:

• A und B einigen sich auf eine elliptische Kurve E über einem end-

lichen Körper Fp, so dass das diskrete Logarithmus-Problem schwer

über E(Fp) zu lösen ist (Diese Eigenschaft wird später noch genauer

betrachtet).

• A wählt ein geheimes a ∈ Z, berechnet Pa = aP und sendet Pa an B.

• B wählt ein geheimes b ∈ Z, berechnet Pb = bP und sendet Pb an A.

• A berechnet aPb = abP . B berechnet bPa = baP .

• A und B einigen sich öffentlich auf eine Methode, aus abP einen Schlüssel

zu generieren, beispielsweise die letzten k-Bits der x-Koordinate von

abP .

Ein potentieller Angreifer (Mithörer) hat die Informationen über die Kurve

E, den endlichen Körper Fp, die Punkte P, aP, bP und die Methode zur

Schlüsselbestimmung. Um den Punkt abP berechnen zu können, muss aber

zunächst a oder b berechnet werden. Sei beispielsweise aP = (xk, yk). Dann

nennen wir das Problem:

aP ≡ (xk, xy) (mod p)

das diskrete Logarithmus-Problem (DL-Problem) über der additiven

Gruppe E(Fp). Da Fp extra so gewählt ist, dass dieses Problem
”
schwer“

zu lösen ist, hat der Angreifer nach heutigem Wissensstand keine effiziente

Möglichkeit, dieses a zu berechnen.

Zur Entscheidung, ob ein DL-Problem über Fp schwer zu lösen ist, wollen wir

im folgenden Abschnitt einen Algorithmus betrachten und implementieren,

der sich dieses Problems annimmt:

7.3.1 Ein Algorithmus für das diskrete Logarithmus-Problem über

additiven Gruppen

Um mit elliptische Kurven rechnen zu können, müssen wir die Addition (Mul-

tiplikation) zunächst in Maxima implementieren.

Das Einzige was hierbei zusätzlich zur oben angegebene Konstruktionsvor-

schrift für die Addition (Multiplikation) beachtet werden muss, ist, dass das

Ergebnis noch in ein Element aus Fp überführt werden muss.

Der Algorithmus folgt auf der nächsten Seite.

(%i136) add ell(E,L,p) := block([m,x1,x2,x3,y1,y2,y3],

if is(L[1] = inf) or is(L[1] = 0) then

return(L[2]),

if is(L[2] = inf) or is(L[2] = 0) then

return(L[1]),

[x1,y1] : mod([L[1][1],L[1][2]],p),

[x2,y2] : mod([L[2][1],L[2][2]],p),

if x1 # x2 then (

m : (y2-y1)/(x2-x1),

x3 : m^2-x1-x2,

y3 : m·(x1-x3)-y1
)

else if is([x1,y1] = [x2,y2]) and is(y1 # 0) then (

m : (3·x1^2+coeff(rhs(E),x))/2·y1,
x3 : m^2-2·x1,
y3 : m·(x1-x3)-x1

) else return(inf),

x3 : mod(num(x3)·inv mod(denom(x3),p),p),

y3 : mod(num(y3)·inv mod(denom(y3),p),p),

return([x3,y3])

)$

(%i137) mult ell(E,P,k,p) := block([i,Q : P],

for i : 1 thru (k-1) do

Q : add ell(E,[Q,P],p),

return(Q)

)$

Die Funktion rhs(E) liefert die rechte Seite der Gleichung der elliptischen

Kurve.

Die Funktionen num und denom liefern den Zähler resp. den Nenner einer

Zahl.

Folgendes Beispiel demonstriert die Arbeitsweise der Algorithmen:

(%i138) E1 : y^2 = x^3 + 2·x + 1$

(%i139) P : [0,1]$

(%i140) Q : [30,40]$

(%i141) add ell(E1,[P,P],41);

(%o141) [1,39]

(%i142) add ell(E1,[P,Q],41);

(%o142) [9,16]

(%i143) mult ell(E1,P,4,41);

(%o143) [38,38]

(%i144) mult ell(E1,Q,8,41);

(%o144) [19,16]

Durch die implementierte Addition & Multiplikation können wir uns nun

dem Algorithmus nähern, vorher aber noch ein wichtiger Satz aus der Alge-

bra:

Satz 7.12. (Satz von Hasse) Die Gruppenordnung (Menge der Punk-

te) einer elliptischen Kurve E(Fp) über dem endlichen Körper Fp kann ab-

geschätzt werden durch:

|N − (p+ 1)| ≤ 2
√
p

wobei N := ord(E(Fp)) =
”
Anzahl der Punkte“.

Dieser Satz ist essentiell für die Implementation des folgenden Algorithmus:

Definition 7.13. (Baby Step, Giant Step) Sei E(Fp) eine elliptische

Kurve über dem endlichen Körper Fp. Ferner seien P,Q ∈ E(Fp) und

N = ord(E(Fp)) die Gruppenordnung von E(Fp).

Wir wollen das Problem Q ≡ kP (mod p) lösen und definieren dazu folgenden

Algorithmus:

1. Wähle ein m ≥
√
N und berechne mP .

2. Erstelle eine Liste der Werte iP für 0 ≤ i < m.

3. Berechne die Punkte Q− jmP für 0 ≤ j < m, bis eines der Resultate

in der Liste aus 2. auftaucht.

4. Falls iP = Q− jmP ist, dann ist Q = kP mit k ≡ i+ jm (modN).

(%i145) B G Step(E,P,Q,p) := block([i,j : 1,L : [],m,

match : false,N,tmpP,tmpQ],

N : ceiling(2·sqrt(p)+(p+1)),
m : ceiling(sqrt(N)),

for i : 1 thru (m-1) do

L : append(L,[mult ell(E,P,i,p)]),

while not match do (

tmpP : mult ell(E,P,j·m,p),
tmpP : [first(tmpP),-second(tmpP)],

tmpQ : add ell(E,[Q,tmpP],p),

if member(tmpQ,L) then

match : true,

j : j+1

),

i : sublist indices(L, lambda([x], x = tmpQ)),

return(mod(i[1] + (j-1)·m,N))
)$

Die Funktionen first und second liefern jeweils das Erste resp. das Zweite

Element einer Liste zurück. Die Funktion member(ELEMENT, LIST) über-

prüft, ob ELEMENT in LISTE vorhanden ist.

Nun wollen wir das Problem Q ≡ kP (mod 41) für die Punkte P,Q und die

elliptische Kurve E aus unserem Beispiel lösen:

(%i146) B G Step(E1,P,Q,41)

(%o146) 23

(%i147) mult ell(E1,P,23,41);

(%o147) [30,40]

Der Algorithmus hat also richtig gearbeitet und als Lösung k = 23 ermittelt.

Was heißt das für unsere kryptographische Anwendung? Viele der Algo-

rithmen zur Lösung des direkten Logarithmus-Problems basieren auf der

Gruppenordnung E(Fp). Für den Anfang ist es auf jeden schonmal wich-

tig, Fp so zu wählen, dass E(Fp) möglichst viele Punkte enthält. Natürlich

existieren noch eine Reihe an weiteren Algorithmen, die nicht zwingend von

einer großen Gruppenordnung behindert werden., deswegen müssen in der

Praxis definitiv noch zusätzliche Sicherheitsmaßnahmen getroffen werden.

69



8 ABSCHLUSSWORT

8 Abschlusswort

Auch wenn in dieser Facharbeit einige Konzepte von Maxima vorgestellt wer-

den konnten, so konnte im Gesamtkontext CAS Maxima doch nur an der

Oberfläche gekratzt werden. Die zahlreichen Funktionen, Zusatzpakete und

Möglichkeiten der direkten Common-Lisp-Programmierung würden aber eine

Facharbeit in diesem Ausmaß bei Weitem übersteigen. Abschließend bleibt

dem Leser nur mitzugeben, wenn durch diese Facharbeit Interesse in der

Programmierung mit Maxima geweckt wurde, sich selbst durch die offizielle

Maxima-Dokumentation tiefere Einblicke zu verschaffen und bei Bedarf auch

aktiv in der Entwicklergemeinschaft Maxima zu partizipieren.

Vielen Dank für das Lesen meiner Bachelorarbeit!

